ALAN Adventure Language Manual

version 2.6

1	INTRODUCTION	7
2	TUTORIAL	9
2.1	What is an Adventure?	9
2.2	Elements of Adventures	
2.3 2.3.1 2.3.2 2.3.3 2.3.4	Alan Fundamentals The Map The Objects The Verbs The Actors	
2.4 2.4.1 2.4.2 2.4.3 2.4.4 2.4.5	Alan Language Description The LOCATION The OBJECT The ACTOR The VERB The SYNTAX	
3	THE LANGUAGE	
3.1	An Adventure	21
3.2	Options	
3.3	Default Attributes	
3.4	Synonyms	24
3.5	Messages	25
3.6	Syntax Definitions	25
3.7 3.7.1 3.7.2	Verbs Verb Alternative Verb Qualification	
3.8	Locations	31
3.9	Objects	
3.10	Containers	
3.11	Events	
3.12	Actors	
3.13	Rules	41
3.14	Start section	41
3.15 3.15.1 3.15.2 3.15.3 3.15.4 3.15.5 3.15.6	Statements Output Statements Special Statements Manipulation Statements Event Statements Assignment Statements If Statement	42 43 45 46 47
3.15.7	Actor Statements	

3.16	WHERE specifications48
3.17	WHAT specifications49
3.18 3.18.1 3.18.2 3.18.3 3.18.4 3.18.5 3.18.6 3.18.7	Expressions50Types of expressions50Logical expressions51The whereabouts of an entity51Binary and Relational operators51The value of Attributes52Literal values52Aggregates52
4	LEXICAL DEFINITIONS
4.1	Comments55
4.2	Identifiers and Names55
4.3	Numbers56
4.4	Strings
4.5	Files
5	EXECUTION OF AN ADVENTURE
5.1	A Turn of Events59
5.2	Player Input59
5.3	Run-time Contexts60
5.4	Moving Actors61
6	HINTS AND TIPS63
6.1	Use of Attributes63
6.2	Descriptions64
6.3	Common Verbs65
6.4	Doors
6.5	Containers
6.6	Imaginary Objects67
6.7	Actors
6.8	Distant Events70
6.9	Questions and Answers70
6.10	Structure71
6.11	Debugging71
7	ADVENTURE CONSTRUCTION75
7.1	Getting an Idea75

7.2	Elaborating the Story	75
7.3	Implementing it	76
7.4	Polishing the Adventure	76
Α	Appendix: RUN-TIME MESSAGES	77
A.1	System Errors	77
A.2	Input Responses	78
A.3	Message Identifiers	81
B	Appendix: ALAN LANGUAGE GRAMMAR .	85
С	Appendix: COMPILER ERROR MESSAGES	93
D	Appendix: HOW TO USE THE SYSTEM	101
D.1	Compiling	101
D.2	Compiler Switches	101
D.3	Running the Adventure	102
D.4	Interpreter Switches	102
E	Appendix: SYSTEM DETAILS	103
E.1	Amiga	103
E.2	Unix	103
E.3	MS-DOS and OS/2	104
E.4	Portability of Games	104
F	Appendix: DIFFERENCES BETWEEN VER- SIONS 105	
G	Appendix: FUTURE DEVELOPMENTS	109
Η	Appendix: REFERENCES	111
Ι	Appendix: EXAMPLE ADVENTURE	115
J	Appendix: COPYING CONDITIONS	119

J.1	Documentation	
J.2	Executables	
J.3	Source	
J.4	Examples	
J.5	Versions, compatibility and support	
J.6	Distribution	
J. 7	Executive Summary	120

1 INTRODUCTION

Text adventures or, using a more appropriate term, interactive fiction is a form of computer game which has many things in common with fiction in book form, role-playing games and puzzle-solving. To create a high quality interactive fiction game, you need to be more of an author than a games programmer.

Alan is a special purpose computer language specifically designed to make it very easy to create such adventure games without any programming skills.

The main principle of the design of the language is simplicity. That is, to make it very easy to do normal things, but also allow more complex things using more complex language constructs. This means that wherever a construct is optional, the system supplies some sensible default instead.

The Alan language has been designed by the author and a very good friend during several years of incremental improvement. This means that the language has a sound foundation, based on practical use, a concept forgotten in many software projects today. Tools develop and are made sharper and more powerful as usage is intensified, the problem domain more and more understood and the requirements increased to tackle new aspects of the problem.

This, we believe, is how software tools must be developed to give the support intended. Therefore Alan and its support system will also develop further. This version is, however, a complete and powerful tool as it stands.

2 TUTORIAL

2.1 What is an Adventure?

As long as man has been around there has been stories, fairy tales and fantasies. In early days the stories were told by story tellers to silent and astonished audiences. After Gutenberg, the stories were printed and the readers partook in the fantasies of the author. In our days, passive viewers are fed from the silver screen or through the tube.

In our century, at last, there has evolved a way for the "audience" to take part in the story themselves. It started in the forties and fifties and continued to develop into the games today known as Dungeon and Dragons, Tunnels and Trolls, etc. Games where a game leader designs the story, but the players decide (and perform) the actions of the characters in the story.

These games, of course, have a computerized counterpart.

The games are played interacting with the computer. The program describes a scene or situation (usually in text, but pictures are more and more frequently used), the player decides on some action and gives orders to the computer to carry out his wishes. Usually there are objects to manipulate, traps to negotiate and puzzles to solve, the object being to find the hidden treasures or save the world.

This form of games was started by Crowther & Woods in the late sixties when they designed the famous *Colossal Cave Adventure* now available on many mainframe computer systems. Inspired by this, Lebling et.al. (then at MIT) took a giant step forward in adventuring by creating the Great Underground Empire and making it available for venturing Adventurers in the game Dungeon. This game contained a much more developed story and could handle much more complex commands.

Later, Dave Lebling & Co started Infocom, a company where they continued to develop their technique, first with *Zork I*, *II* and *III* (the first a reimplementation of Dungeon, the others equally successful sequels). Since then a host of games has been released (*Starcross, Witness, Enchanter* are some of the names that come to mind). Infocom today only exists as a label with Mediagenic, the original authors scattered, but the Infocom games are still among the best available today.

Other companies have followed in Infocom's footsteps and a handful of them seem to make a living out of creating adventure games.

There have been many attempts to use computer graphics to display the surroundings and objects in adventure games. Some of the more successful are the Sierra OnLine games (notably the *Leisure Suit Larry* and the *Kings Quest* series) which have mouse oriented moves but also allows single line text commands, games from ICOM Simulations (*DejaVu* and *The Uninvited*) which are purely graphics games with mouse and icon interfaces. Other manufacturers have tried to use (some times optional) pictures to accompany the text, for example Magnetic Scrolls games (e.g. *the Pawn*), which shift the picture automatically as you move around using the normal directional commands.

A large number of so called Arcade Adventures are also available but they are always more Arcade than Adventure and do not belong in this discussion.

The Alan Adventure Language is designed to aid construction of pure text adventures or, in the words of Infocom, interactive fiction. Possibly, in future versions sound and graphics functions may be available.

2.2 Elements of Adventures

The success of all Infocom games can probably be attributed to three distinctive features. First, they all have a 'believable' and consistent plot, which is flavoured with humour and wittiness. Second, the descriptions are extensive and give a lot of atmosphere to the game. Third, the command handler recognizes and understands a large vocabulary and complex input. Add to this the worlds best graphics device (the human brain) and you are unbeatable!

Looking at Adventures in more detail, one sees some common features. There is, of course, a world or universe (called the map) where the Adventure is taking place. Although you can move around quite freely there are usually some problems getting into certain parts of the world (like locked doors, no air to breathe or even finding the entrance).

The size of the map ranges from hundreds of locations (like in some of the Infocom games) to just two or three (some of the Scott Adams games are very compact and very difficult).

Then, there are the objects in the game. These range from your tools, like lamps and shovels, to immaterial things like a hole in the ground, in short, anything you can manipulate. Ideally everything that is mentioned in a description should be an object, but this is normally impossible because of storage limits (and perhaps the stamina of the games designer!).

Most objects have uses. A key is easy to guess how to use, but what about the velvet pillow? Red herring objects are also common in adventuring.

The player must be able to express his wishes. Natural language commands, advocated in many advertisements, are probably overkill since most players do not wish to wear their fingers out typing full sentences, but single verb-object input is not sufficient for a good game either. The player must be able to say things like

> take all except the blue vase

or

> put the ring and the bag in the box

2.3 Alan Fundamentals

The Alan Adventure Language is a high level computer language designed to make it easy to create text Adventures. The Alan system handles all the tiresome tasks and supplies reasonable defaults so you can concentrate on the plot, the puzzles, the objects and the map.

There are no variables, subroutines or other traditional programming constructs in Alan because Alan is not a *programming* language. Instead Alan takes a descriptive view of the concepts of adventure authoring. The Alan language contains construct to make it possible for the author to describe the various features of these concepts. By describing for example how the locations in the adventure are connected you have described the geography in which the story will take place. Much of what should be described is in terms of text that should be output to make the player experience the story that you have designed.

Another fundament of Alan is that the execution of an adventure is event driven. This means that the things happening in the adventure are triggered by one thing only, namely the input of a player command. This command is analysed according to the allowed syntax and transformed into execution of verbs or movements which in turn executes other part of the description in the Alan source. After a player turn other actors can move and scheduled events can be run, then the player takes another turn.

2.3.1 The Map

In Alan, the map is a number of locations connected (or not!) by any number of exits. A location has a description that is presented to the player when it is entered. A location may also have a number of exits stating in which direction the exit is and to which location it leads. Alan places no restrictions on the layout of the map, any topology is legal. Note, however, that since exits are oneway, an explicit declaration of a backward path (if such is desired) must be made.

2.3.2 The Objects

The other vital entity in Alan is the object. Most Alan objects are things that in real life would be objects too, like a knife or a key. In addition, other things that should be manipulable by the player, e.g. parts of the scenery, must be declared as an object. For example if you require the player to 'whistle the melody', then the melody must be an Alan object.

Objects, like locations, have a description that is presented when they are encountered during the game.

Every object may also have a set of properties, like eatable and movable, which may be changed during the execution of an Alan program. Most object would probably not be eatable so there is also a mechanism for giving default properties to objects. Some verbs have special meaning or effects when applied to a certain object. These verbs and their effects are also declared within the object.

2.3.3 The Verbs

Verbs are imperative statements input by the player to command some action. These must also be declared in an Alan adventure, either in an object (as described above) or as a general (global) verb that describes the effects of the verb when applied to any object or for verbs to which no object may be given.

To make it possible for the player to input more complex commands a means to specify the syntax for a verb is also available. Using this verbs can also be made to operate on literals (strings and integers) giving the player the possibility to input things like

> write "Merry Christmas, Mr. Lawrence" on the xmas card

2.3.4 The Actors

An extra thrill and dimension are additional characters in the game. These are called actors and have a life of their own. For each move the player makes these programmed characters also get a turn to do their thing. An actor may be a thief running around and stealing your collected treasures, a dragon guarding the entrance to its lair and so on.

One of the interesting things about playing adventure games with actors is to figure out how to interact with and influence the other characters.

2.4 Alan Language Description

Alan is an Adventure language, i.e. a language designed to make it easy to write Adventures. This means that the Alan language must reflect the various entities encountered when creating an Adventure plot.

The first step after having come up with a plot for your Adventure is to draw a map of the world where the Adventure is taking place. For this purpose the LOCATION construct is provided.

Next step is to introduce tools, weapons and other manipulable objects. Here the OBJECT construct is used.

Then you need words to command action. The language construct to supply these with is the VERB. You may also define more complex types of player input using the SYNTAX construct.

Additionally, you may also want other characters and creatures in your adventure. For this the ACTOR construct is provided.

2.4.1 The LOCATION

The scene for your Adventure is a series of "rooms" or, rather, locations. They are connected by paths that lead from one location to another, making it possible for the hero to travel through the world of your design, exploring it and solving the puzzles.

What is necessary to describe a location? Every location (or LOCATION, when we are referring to the Alan language construct) must have a name. This is so that you, the designer, may refer to that LOCATION easily, instead of having to remember a magic number for each LOCATION.

Unless you provide other means for transportation from a LOCATION, you should also describe in which directions there are exits and to which LOCA-TIONs they lead.

In fact, this is all that is really necessary in a LOCATION, so lets look at an example (you would probably like to try this out, referring to appendix E, *SYSTEM DETAILS*, on page 103 for instructions for your particular system).

```
LOCATION Kitchen
EXIT east TO hallway.
END Kitchen.
LOCATION Hallway
EXIT west TO kitchen
END Hallway.
START AT Kitchen.
```

This is a complete Alan Adventure (although very primitive). As you see, every Alan construct ends with a period ('.') and there is also a "START AT" sentence at the end, indicating in which location to put the hero when the game starts.

Run this little Alan source through the Alan Compiler (see appendix D, *HOW TO USE THE SYSTEM*, on page 101 and appendix E, *SYSTEM DETAILS*, on page 103) and try the Adventure. After starting the Adventure, two lines will be shown on your screen. The first line will contain "Kitchen" and the second a ">", which is the prompt for the player to input a command. Now try typing "east". The word "Hallway" and the prompt will appear. Typing "west" will take you back to "Kitchen" again.

We see that the name of a LOCATION is automatically used as a description shown when that room is entered. We also see that the words used in the EXIT-parts are translated by Alan into directional commands that are usable by the player.

You should remember that exits are strictly one-way. An EXIT from a LO-CATION to another does not automatically imply the opposite path. Thus one must explicitly declare the path back, in the definition of the other LOCA-TION. But just the name of the location is not much of a description. So in order to provide the "purple prose" descriptions often found in high-class Adventures there is an optional DESCRIPTION-clause in the LOCATION. Let us describe the Hallway.

```
LOCATION Hallway
DESCRIPTION
"In front of you is a long hallway. In one end
is the front door, in the other a doorway. From
the smell of things the doorway leads to the
Kitchen."
EXIT west TO Kitchen.
END Hallway.
```

We introduce another feature in this example, namely the text enclosed in double quotation marks ('"') which is called a STRING or, when used on its own like this, an output statement. When executed this string will be presented to the player and formatted to suit the format of his screen.

Invent a description for the Kitchen, enter it in the Alan source and run the changed adventure. You notice, of course, that the text in the output statements is reformatted during output to suit your screen, in order to make room for as much text as possible. Note also that you do not have to worry about this at all - in your source file you may format the text any way you like.

This type of output statement is just one of the statements in the Alan Language, and we will see more of them later.

It is also possible to have conditions and statements in the EXIT-clauses of a LOCATION to restrict the access to the next location or to describe what happens during this movement.

```
EXIT west To Kitchen
CHECK kitchen_door IS open
ELSE "The door is closed."
DOES
"As you enter the kitchen the smell of
something burning is getting stronger."
END EXIT west.
```

2.4.2 The OBJECT

Another essential feature in Alan is the OBJECT. Like the LOCATION, the OBJECT is a means to describe the "physical" world where your Adventure is taking place. Most objects are probably used to provide puzzles, like closed doors, keys and so on, but other objects should be promoted to OBJECTs too. A large number of objects that can be examined and manipulated makes a game so much more enjoyable.

OBJECTS, like LOCATIONS, have names and descriptions, so you might guess the general structure of an OBJECT.

```
OBJECT door AT Hallway
IS closed.
DESCRIPTION
```

```
"The door to the kitchen is a sliding door."
IF door IS closed THEN
"It is closed."
ELSE
"It is open."
END IF.
END OBJECT door.
```

An OBJECT may initially be located at a particular LOCATION (although objects do not have to start at a particular place in which case they are not present in the game until located at some place where the player may lay his hands on them). This is indicated by the AT-clause, in this case telling us that the door is initially located in the Hallway.

In addition, OBJECTs may have attributes indicating the state of certain properties of the object. In this example with a door, the IS closed part indicates that the OBJECT door should initially have the attribute closed set to TRUE (i.e. the door is initially closed). The opposite would be indicated with a NOT, (i.e. IS NOT closed).

Alternatively, attributes may be numeric (e.g. HAS weight 5) or be of string type (e.g HAS inscription "Kilroy was here").

We also introduce another Alan statement, the IF statement. The IF statement allows you to select which statements to execute according to some condition. In the example, the closed attribute of the door selects which description to show. There are further variations of expressions and the IF statement, but we will come back to these later (*Expressions* on page 50 and *If Statement* on page 48).

Instead let's look at some other statements in relation to OBJECTs.

The attributes of an OBJECT must, of course, be changeable, and this is done with the MAKE statement or the SET statement. For example if the door should be opened (the player having said "open door", perhaps) this could be performed by

MAKE door NOT closed.

or closed (i.e. setting the closed attribute to TRUE again) by

MAKE door closed.

The SET statement changes numeric or string attributes, for example

SET level OF bottle TO 4.

Another OBJECT manipulating statement is the LOCATE statement. This is the statement to use when moving objects from one location to another. Opening a lid might cause a previously hidden object to fall to the floor, something that could be performed by moving the object from a limbo LOCATION to the current one with

LOCATE treasure HERE.

Or to a particular place with

LOCATE vase AT hallway.

2.4.3 The ACTOR

The ACTOR is used to populate the adventure with other creatures. They might be pirates or monsters, but the thing they have in common is that they move around and perform various actions more or less in the same way as the player does.

An ACTOR may have a DESCRIPTION and attributes like OBJECTs and LOCATIONS. An ACTOR performs his movements by following scripts, each having a number of steps. Each step corresponds to one player move.

```
ACTOR charlie_chaplin NAME Charlie Chaplin
SCRIPT 1
STEP
LOCATE ACTOR AT outside_house.
STEP
LOCATE ACTOR AT hallway.
USE SCRIPT 1.
END ACTOR charlie_chaplin.
```

2.4.4 The VERB

The VERB is the construct that implements the effects of an action requested by the player. VERBs may be global, local to a particular LOCATION or associated with an OBJECT. We will look at the implications of various combinations of these in the next few sections.

To implement a VERB you need a name for it (which is also the default word the player should input to request that action). You must also decide which effects this verb should have under various circumstances.

If we want to implement the VERB open for the door we could use the following code

VERB open DOES MAKE door NOT closed. END VERB open.

This implementation makes direct references to the door, so to make the verb more general it would instead need to reference the object the player mentioned in his command (see *The SYNTAX* on page 19 for a discussion on this). In this case the attribute closed must also be available for all objects (by making it a default object attribute (see *Default Attributes* on page 23).

Of course, there are also conditions that have to be checked before we could execute this code (perhaps to see if it was possible to open the object!). Therefore VERBs may have CHECKs.

Checking Things

In order to assert that the correct conditions are fulfilled before a VERB is actually executed the VERB has an optional CHECK part.

```
VERB open
CHECK OBJECT IS openable
ELSE "You can't open the $o."
DOES
MAKE OBJECT NOT closed.
END VERB open.
```

This is a more probable definition of the open VERB than the previous one. What it means is that before the statements after DOES are executed, the condition after CHECK must be checked (that the object indicated by the player is really openable). If the condition is TRUE then the requirements are fulfilled and the body of the VERB may be executed. If this is not the case, the ELSE part is executed instead (normally some error message).

A CHECK may have multiple conditions as the following code shows:

```
VERB take
CHECK OBJECT takeable
ELSE "You can't take that."
AND OBJECT NOT IN inventory
ELSE "You already have it."
DOES
LOCATE OBJECT IN inventory.
END VERB take.
```

Here we encounter a variation on the LOCATE statement - the capability to place an object inside a container.

Global, Local and Objective VERBs

VERBs may be defined on three levels.

- Globally. These are always used, no matter in what location the player currently is, or what object he is trying to manipulate.
- Locally (within a particular LOCATION). A local VERB is only considered when the player issues the VERB at a particular LO-CATION.
- Within an object. When the player tries to manipulate the object within which the VERB is defined, the VERB definition in that OBJECT is executed.

A VERB may be defined on all three levels (as well as in other LOCATIONs and OBJECTs of course), and may have CHECKs in all instances. The implication is that all CHECKs must be passed before any execution and if they all do pass the verb bodies (DOES parts) are executed. The order is global/local/object.

An example:

```
VERB throw
  CHECK OBJECT IN inventory ...
  DOES
    LOCATE OBJECT HERE.
END VERB throw.
LOCATION dark place
  VERB throw
    CHECK "Too dark to aim."
  END VERB.
END LOCATION dark_place.
OBJECT vase
  VERB throw
    DOES
       "The vase breaks."
       LOCATE vase AT limbo.
  END VERB throw.
END OBJECT vase.
```

The CHECK without a condition in dark_place is called unconditional and is always FALSE (i.e. it will always fall out as if the CHECK had failed).

Now assume the player is carrying the vase at dark_place. He says

> throw vase

So we have a VERB globally as well as locally and in the mentioned object. The CHECKs are examined in the following order:

- OBJECT IN inventory? (in the global VERB)
- unconditional (in the LOCATION)
- None (in the OBJECT)

We fall out already in the LOCATION (player receiving the response "Too dark to aim.") so the third (empty) CHECK is never tested. Now the player tries the same thing at bright_place where there is no restriction on throwing (no local VERB "throw").

This time there is no local VERB so we skip that level and get the CHECKs

- OBJECT IN inventory? (in the global VERB)
- None (in the OBJECT)

Each is tried in turn and none fail, so we can go ahead and execute. This is done in the same order, i.e.

- LOCATE OBJECT HERE (in the global VERB)
- nothing (in the LOCATION)
- "The vase breaks..." (in the OBJECT)

Note: You can never destroy an OBJECT or remove it from the game. Instead, you will probably need a limbo location, i.e. a location that is not connected to any of the others and may thus be used as a storage for destroyed objects and other things the player is not supposed to see.

2.4.5 The SYNTAX

Normally a verb acts on one object or actor, henceforth called a parameter. This means that the format of player input normally is something like

```
> take vase
```

This form, or syntax, is the default form when you don't specify anything else. The default syntax might thus be described as

```
SYNTAX
? = ? (object)
WHERE object ISA OBJECT.
```

where the question marks are replaced by the name of the verb.

But in order to allow different and more complex player input the SYNTAX construct is supplied.

The SYNTAX construct is a way to describe the words and parameters the player may use in order to execute a particular verb (its global and more specialised parts). Below is the syntax for put_in, the verb to put something inside a container.

```
SYNTAX
   put_in = 'put' (obj) 'in' (cont).
```

This syntax defines the put_in verb to be executed when the player has input the word 'put' followed by a reference to an object or actor (a parameter named obj), followed by the word 'in' followed by a reference to a second parameter (the container), as in

> put the green pearl in the black box

This will bind the parameter obj to the object that represents the green pearl and the parameter cont to the black box. It is also possible to restrict the types of the parameters:

```
SYNTAX
  put_in = 'put' (obj) 'in' (cont)
  WHERE obj ISA OBJECT
    ELSE "You can't put that into anything."
    AND cont ISA CONTAINER OBJECT
    ELSE "Nothing fits inside that."
```

This restricts the parameter obj to being an object (as opposed to an actor for example) and the parameter cont to a container object (an object with the container property).

19

The parameters are used as normal identifiers in the Alan source, provided they are defined in the current context, i.e. they can only be used in the bodies of the verb (see also *Run-time Contexts* on page 60 for a detailed discussion).

The SYNTAX construct generalises the verb execution order described previously from execution of verbs in one object, to verb bodies in all the parameters. In the example above, verb bodies in both the object or actor referenced as obj and cont (the green pearl and the black box) are executed (if present in their declarations).

Another use for the SYNTAX is to define the syntax for simple verbs such as quit, score etc. They also need a SYNTAX definition as they do not fit into the default verb/object format. An example would be

SYNTAX q = 'quit'.

But for simple verb/object forms no SYNTAX is actually necessary.

Note: In expressions, OBJECT always refers to the first parameter. This makes it consistent with the default syntax of verb/object (and also with the definition of OBJECT in version 1).

3 THE LANGUAGE

This chapter describes the Alan language in detail. For each construct the syntax is given in the form of BNF productions. A description of this notation may be found in appendix B, *ALAN LANGUAGE GRAMMAR*, on page 85. The numbering of rules is also the same as in that appendix.

3.1 An Adventure

```
adventure = optional_options units start
;
```

An adventure starts with an optional options section and then a set of units.

```
9. units = unit
    ! units unit
10.
          ;
11. unit = object_default
       ! location_default
12.
13.
        ! actor_default
14.
        ! messages
15.
        ! rule
        ! synonyms
! syntax
16.
17.
18.
        ! verb
19.
        ! location
20.
         ! object
21.
         ! container
22.
         ! actor
23.
         ! event
```

The major part of the adventure is composed of units which are rules, synonyms, syntax definitions, verbs, locations, objects, containers, actors and events that can be declared in any order. Any combination and number are allowed. Also default attributes for objects, locations and actors may be declared in any number of places.

149. start = 'START' where '.' optional_statements

The adventure source text must end with a start section. It indicates where the hero is when the game starts and can be used to set things up, welcome the player and so on. The start section is mandatory.

```
START AT bedroom.
SCHEDULE alarm_clock AFTER 2.
"Slowly you come to your senses, your numb limbs
starting to feel the blood flowing through them..."
```

3.2 Options

2. 3.	optional	 options		'OPTIONS'	options
4. 5.	options	option options	5 C	option	

; 6. option = ID '.' 7. ! ID ID '.' 8. ! ID INTEGER '.'

Options define things concerning the overall behaviour of the generated Alan adventure. An option is for example written either as

LANGUAGE Swedish.

(for multiple-valued options) or

PACK. NOPACK.

(for boolean options).

The options are

Option name	Values	<u>Default</u>
Language	English Swedish ^a	English
Width	24-255	80
Length	5-255	24
Pack	Boolean	Off
Debug	Boolean	Off

a. Other non-English languages may be supported in the future depending on demand.

The Language option specifies in which language the adventure is intended to be played, and selects different default message texts. Alan is primarily designed for adventures in the English language, but it is also possible to write adventures in other languages. To make this possible, the default messages output by the interpreter may be generated in different languages.

The Alan compiler and interpreter will always allow multinational 8-bit characters as input and the default messages is generated for 8-bit character sets, internally representing national characters according to the ISO multinational character set (ISO8859-1) requiring 8 bits. On output this is converted to the native character set of the machine (whenever possible) which means that portability between platforms should be good even for text containing non-ASCII characters.

Width specifies how long the lines the interpreter outputs should be (formatting is automatic!). The Length option will instruct the interpreter to how many lines to show on the screen without any player interaction (<More>). *Note:* In some environments the Width and Length options may be overridden by the current values of the screen or window if the operating system can supply them.

The Pack option will cause the compiler to compress the texts to occupy less space. As a bonus this also makes it impossible for the player to cheat by dumping the adventure text data file. As a drawback it does make execution of the adventure a bit slower (quite noticeable on some smaller computers).

In order to allow debugging of the generated adventure (see *Debugging* on page 71), the debug option must be turned on. This may also be performed using the debug compiler flag (see also appendix D.2, *Compiler Switches*, on page 101).

3.3 Default Attributes

An attribute is a definition of a property of either an object, an actor or a location. An attribute can be boolean (having the value TRUE or FALSE), numeric or of string type. The type of an attribute is inferred from the type of its initial value. Attributes that every object, actor or location has by default should be declared in the OBJECT/ACTOR/LOCATION ATTRIBUTES sections. An boolean attribute is declared either as

attribute_name. NOT attribute_name.

or as

attribute_name INTEGER.

for boolean and numeric attributes, respectively or as a string typed attribute by

attribute_name STRING.

For example

ACTOR ATTRIBUTES NOT hungry. weight 50. Note that string valued attributes are mainly intended for saving string parameters from the player input, like in

> scribble "Kilroy was here" on the wall

It is not intended for keeping long strings of descriptions, especially not as default attributes as they (in the current implementation) require much space and takes long time to initialise when starting the game.

Any number of default attributes sections are allowed. This makes it possible to group verb declarations (see below) and the declaration of the default attributes that particular verb requires.

3.4 Synonyms

Synonyms declare *words* that, when used as player input, are interchangeable at all times. The word on the right hand side of the equal sign must be a word defined elsewhere in the adventure source, such as an object name (a noun) or a direction. The list of words on the left hand side are new words (*NOT* defined elsewhere) that always will be replaced by the word on the left in the player input.

```
SYNONYMS
'i', 'invent' = 'inventory'.
'q' = 'quit'.
```

When defining synonyms remember that this only defines player *words* that are interchangeable. Defining synonyms for verb names etc. will not always give you the result that you expect. For example

```
SYNONYMS
'examine' = look_at.
SYNTAX
look_at = 'look' 'at' (obj).
VERB look_at ...
```

This will result in the compiler issuing an error message indicating that the synonym word 'look_at' is not defined. This is because the SYNTAX (se below) defined the *verb* look_at to have the specified syntax (including the player words 'look' and 'at'), the player word 'look_at' is not defined, which is as well as the player would not be able to input a word with an underscore (see *Player Input* on page 59).

3.5 Messages

The Alan system has a number of standard messages built in. These messages are presented to the player in various situations, both normal and otherwise. An example is the following:

```
> go north
You can't go that way.
```

The response "You can't go that way." is a typical example of such system messages (see also appendix C, *COMPILER ERROR MESSAGES*, on page 93).

To make the user dialogue more adapted to the settings you select Alan allows you to define your own versions of the messages. The syntax to do that is shown above, and below is an example:

```
MESSAGE
NOWAY: "There is no exit in that direction."
```

If the above is used in the source for same game as the previous example, it would instead look like:

```
> go north
There is no exit in that direction.
```

The standard message is replaced by the contents of the string in the definition. For a complete list of all the identifiers of messages and their use see appendix A, *RUN-TIME MESSAGES*, on page 77.

3.6 Syntax Definitions

The syntax construct is used to specify the structure of the input from the player. Each syntax_item defines the syntax for one VERB. The effects are declared using the VERB construct (see *Verbs* below).

The syntax is defined as a number of elements each being either a player word or the name of a parameter (an identifier enclosed in parenthesis). The optional_multiple_indicator (an `*') may be used to indicate that this parameter can reference a set of objects or actors (for example by the player using all or concatenating a number of parameters using a conjunction like and).

```
SYNTAX
take = 'take' (obj)*.
drop = 'drop' (obj).
```

This shows the syntax definitions for the verbs take and drop, take also allowing multiple objects. This would allow inputs like

```
> take everything except the pillow
> drop the vase
```

but not

> drop the shovel and the bucket

For more information on player inputs refer to *Player Input* on page 59.

```
51. optional_class_restrictions = '.'
                   ! 'WHERE' class_restrictions
52.
                             ;
53. class_restrictions = class_restriction
54. ! class_restrictions 'AND' class_restriction
55. class_restriction = ID 'ISA' classes 'ELSE' statements
56. classes = class_identifier
  ! classes 'OR' class_identifier
57.
          ;
58. class_identifier = 'OBJECT'
    ! 'ACTOR'
59.
                   ! 'CONTAINER'
60.
                   ! 'INTEGER'
61.
                   ! 'STRING'
62.
                   ! 'CONTAINER' 'OBJECT'
! 'CONTAINER' 'ACTOR'
63.
64
```

To restrict the types of entities the player may refer to in the place of a parameter its class may be defined by using element_classes. Each parameter may be restricted to refer only to certain kinds (classes) of entities: objects, objects with the container property, actors, numeric literals, string literals or a combination of these. The statements following the ELSE will be executed if that restriction is not met, i.e. if the player made a reference to an entity not in the specified class or classes. The default is OBJECT, i.e. if no element_classes are supplied for that parameter identifier the player may only refer to objects at that position in his input.

The following example describes the syntax for a verb which only allows OB-JECTs as its parameters (this is however also the default, see below).

```
SYNTAX
take = 'take' (obj)
WHERE obj ISA OBJECT
ELSE "You can't take that."
```

The classes defined for a parameter are also used by the compiler to analyse statements and expressions in which that parameter occurs to ensure that the entity referenced is guaranteed to have the properties required during runtime. A parameter identifier defined using ISA OBJECT may for example not be used in a LIST statement as this requires the entity to have the container property (ISA CONTAINER would of course restrict the entities to only those entities that are containers and would do the trick). As there are a number of kinds of entities that can have the container property it is possible to restrict parameters to only those objects that are containers (CONTAINER OBJECT), those actors that are containers (CONTAINER ACTOR), or that it need just have the container property. The last case will not allow access to attributes of the parameter, as you can not be sure what type it is.

If no SYNTAX is defined for a VERB at all, that is equivalent to specifying

SYNTAX ? = ? (object).

The question marks represents the name of the VERB. This means that normal verb/object type of VERBs by default have the correct syntax and may only refer to objects. It also implies that the default name for the single parameter is OBJECT (see *WHAT specifications* on page 49 for the implications of this).

Note: Following this default mechanism all verbs that has no corresponding syntax declaration will be assumed to require an object as parameter. This means that simple 'verb-only' VERBs must be declared using a syntax like q = 'q' to make it possible to input a single verb word. It also means that verbs that have no SYNTAX will only accepts OBJECTs, not ACTORs for example.

3.7 Verbs

A verb declaration specifies the effects of something the player does (commands using a syntactically legal input). Verbs can be declared globally, at a location or inside an object or actor.

The identifiers in the id_list will by default also be the player words that can be used to invoke the verb. If a SYNTAX is declared for the VERB, the identifiers in the id_list will not normally be accessible to the player, instead the sequence of words and parameters specified in the SYNTAX must be used.

Checks

76. optional_checks =
77. ! 'CHECK' statements
78. ! 'CHECK' check_list
79. check_list = check
80. ! check_list 'AND' check
;
81. check = expression 'ELSE' statements

To decide if the action is at all possible to carry out, the CHECKs are executed. First the global checks are tried, then the checks in the verb declaration at the current location and finally the checks declared for the verb in the objects or actors bound to the parameters (if any).

Note: The verb need not be declared in all of these places.

If no expression is specified for a check, the check will always fail, an unconditional check. This is useful for preventing certain actions at specific locations for example.

```
LOCATION 1
VERB jump
CHECK "You can't do that here."
END VERB jump.
END LOCATION 1.
```

If any check should fail, the execution of the current verb is interrupted and the statements following the failing check are executed. The user (player) is then prompted for another command.

Does-clause

If all checks succeed the DOES-part(s) of the VERB will be carried out. The order is normally to first execute the body any global declaration, then the body in the verb declaration for the current location. Finally each parameter is examined to find any declarations of the VERB inside what it refers to, those verb bodies are then executed in the order in which the parameters occurred in the syntax declaration. This is the most natural order and covers most cases but in some infrequent situations another order may be necessary. By using the qualifiers BEFORE/AFTER/ONLY the author can decide which verb

bodies will be executed and in which order (see section 3.7.2 below for details).

```
VERB take
CHECK obj NOT IN inventory
ELSE "You already have that."
DOES
LOCATE obj IN inventory.
END VERB take.
```

3.7.1 Verb Alternative

When a VERB is declared inside an OBJECT, verb_alternatives are allowed. These alternatives are used in conjunction with the SYNTAX declaration defined below in the following way. When a player inputs a command each parameter in the syntax (see above) is bound to an actual object or actor or receives the value of a literal, depending on the specified syntax. To find out which verb CHECKs to test and verb bodies to execute the parameters are examined in turn according to the algorithm described in *Verb Qualification* below. Each object may then have a different verb body, depending on at which position it occurred (to which parameter it was bound).

For example with the syntax definition

SYNTAX break_with = 'break' (o) 'with' (w).

the VERB body for break_with to execute for the delicate_vase probably would differ if it occurs as the direct object (o), or if it occurs as the indirect object (w). For each such parameter in the syntax you may define different actions by supplying a verb_alternative for each parameter identifier. The verb declaration could look like

```
OBJECT feather

VERB break_with

WHEN O DOES

"The feather is even more flat than before."

MAKE feather flat.

WHEN w DOES

"There is nothing much that you can break with a

feather!"

END VERB break_with.

END OBJECT feather.
```

3.7.2 Verb Qualification

234.	optional_qual	=	
235.		!	'BEFORE '
236.		!	'AFTER '
237.		!	'ONLY'
		;	

The order in which the different verb definitions are executed is normally from the outside in, i.e. the global definition is executed first if a global definition exists, then any possible definition of this verb in the current location. Lastly, the verb bodies in the parameters (in the order they appeared in the syntax definition) on which the verb was applied (if any) is examined to find and execute their verb definitions.

In most circumstances this is the most logical order, but if another order is required the verb qualifiers AFTER, BEFORE and ONLY may be used to alter this behaviour. The qualifiers alter the order of execution and a strict definition of this is described below.

First, the verb in the last parameter (if any) is investigated and, if this definition had the BEFORE or ONLY qualifier it is executed. If the qualifier was ONLY the execution is also aborted at this stage and no more verb definitions are examined, otherwise the other parameters are examined in the same way.

In the next step, the current location is examined and, if it contained a verb definition with a BEFORE or ONLY qualifier, that definition is now executed (and if it was ONLY, execution is aborted). As a result a BEFORE qualifier in the verb definition in an object will supersede an ONLY qualifier in the location.

At this stage, all BEFORE and ONLY qualifiers are handled appropriately since the global definition is now in turn anyway. This leaves the definitions without any qualifier or with the AFTER qualifier. The global definition is examined and if it did not have the AFTER specification, it is executed (if it had a ONLY qualifier execution is stopped after executing it). Any definition of the verb in the current location is again examined and, if it did not have the AF-TER qualifier, it is executed. What remains is to execute the verb definition in the parameters if they have not been executed already, and to execute the location definition and the global definition (in that order) if they where declared with the AFTER qualifier.

So in short (with global definitions being the outermost and the definition in the entity bound to the last syntax parameter the innermost):

- From the outside in, find any BEFORE or ONLY definitions and execute them (stop if ONLY found).
- From the inside out, execute any definitions not already executed and not declared with the AFTER qualifier.
- Execute the remaining verb definitions (those with an AFTER qualifier) from the outside in.

The normal order of execution is represented by the second item in the above list.

The qualifiers are a powerful but confusing concept. The normal order of execution is usually appropriate and only in special cases should qualifiers be used. When they are needed, you will find that one qualifier at the correct definition will normally do the trick. The above algorithm is used to get a strict definition of the execution order. It is not expected that this complex behaviour will be needed in practice.

Note: All checks for a VERB will always be run in global-location-parameter order regardless of any BEFORE/AFTER/ONLY qualifiers.

An example of the use of qualifiers is to ensure that only the verb body within the object is executed:

```
OBJECT bomb
VERB take
DOES ONLY
"Your curious fingering at the intricate
mechanism sets it of. BOOOM!"
QUIT.
END VERB examine.
END OBJECT bomb.
```

3.8 Locations

A location is a declaration of a place (a "room") in the game that (normally) can be visited by the player, have objects lying around, etc.

Identifier and Name

114. optional_name =
115. ! 'NAME' ids
.

The ID is the identifier used by the author throughout the source when referring to this location. By default, this will also be the name of the location written out to the player, but by using the NAME clause you can give a different name to the location when presenting it to the player (see *Objects* on page 33 and *Identifiers and Names* on page 55).

Attributes

A location can have attributes (see *Default Attributes* on page 23). These can be local attributes available only for this location or override declared default attributes.

Description

The statements in the DESCRIPTION clause should print a description of the location. These statements are executed when the hero enters the location or when executing a LOOK statement. See also *Special Statements* on page 43, concerning the VISITS statement.

Does-clause

The optional DOES clause contains statements performed when *any* actor enters the room (is located there). An example usage of this would be if there was a weak bridge that only allows a certain total weight before it collapsed. The DOES clause of that location could contain checks for this which would be executed whenever any actor enters that location, not only the hero.

Exits

To build a world of locations, these must be connected. This is done by using exits. An exit consist of an id_list, all of which are considered directional words, i.e. when input by the player, they will move him to the location identified by the ID. It is possible to customize the exit using CHECKs (see *Verbs* on page 27 for a definition), that must be satisfied to allow passage through the exit, and statements that will be executed when the player passes through.

Note: If there exist an exit from one location to another, there will *NOT* automatically be an exit in the opposite direction!

Two interconnected locations might be declared like:

```
LOCATION east_end NAME 'East End of Hall'
DESCRIPTION
"This is the east end of a vast hall. Far away to the
west you can see the west end."
EXIT w TO west_end.
```

```
END LOCATION east_end.
LOCATION west_end NAME 'West End of Hall'
DESCRIPTION
   "From this western end of the large hall it is almost
        impossible to discern the opposite end to the
        east."
EXIT e TO east_end.
END LOCATION west_end.
```

Local verbs may also be declared in a LOCATION. See *Verbs* on page 27 for a description of this.

3.9 Objects

```
93. object = object_header object_body object_tail
94. object_header = 'OBJECT' ID optional_name optional_where
95. object_tail = 'END' 'OBJECT' optional_id '.'
             :
;
99. is = 'IS'
100. ! 'ARE'
101. ! 'HAS'
102. optional_descriptions =
                     ! optional_descriptions
103.
                       article_or_mentioned_or_description
104. article_or_mentioned_or_description = article
                                 ! description
105.
106.
                                 ! mentioned
107. optional_description =
108.
      ! description
```

Objects are all the things that can be manipulated by the player. They can be picked up, examined and thrown away (if the author has allowed it). They will usually be described when the player enter a location containing objects.

As for locations, the ID is the name you use to refer to this object. It is also the default name for what is presented to the player and what he has to use when referring to the object.

Name

114. optional_name = 115. ! 'NAME' ids

By using the NAME clause you can give the object an other name, e.g.

OBJECT chair3 NAME little wooden chair

In this example the word "chair" is a noun and "little" and "wooden" would be adjectives. When the player refers to the object with the author name chair3, he may use just "chair" if it is the only object with "chair" as its noun at the current location, or he may distinguish between multiple chairs by also giving one or more adjectives to pin down the chair he wanted.

Note: If the NAME clause is used the name chair 3 is *not* available to the player.

Initial location

It is possible to set the initial location of an object by using an optional where clause. If no such clause is used the object will not be present in the game until it is moved somewhere by a LOCATE statement. Only the AT what and IN what forms are allowed when describing an initial location of an object.

```
OBJECT chest AT tresury
IS NOT open.
DESCRIPTION
  "On the floor there is a heavy golden chest. Its sides
   and top are completely encrusted with jewels."
MENTIONED
   IF chest IS open THEN
      "open"
   END IF.
      "heavy jewel encrusted chest"
   VERB open ...
   :
END OBJECT chest.
```

Properties

An object can also be a container. This is declared by means of the CON-TAINER property clause, which looks like an ordinary container declaration (see *Containers* on page 36).

```
OBJECT chest
CONTAINER
LIMITS ...
HEADER ...
DESCRIPTION ...
:
END OBJECT chest.
```

Attributes

```
27. attributes = attribute '.'
28. ! attributes attribute '.'
;
29. attribute = ID
30. ! 'NOT' ID
31. ! ID optional_minus INTEGER
32. ! ID STRING
.
```

An object can have attributes (see *Default Attributes* on page 23). These can be local attributes or override values of declared default attributes.

Article

The optional article can be used to define the indefinite article that should be placed before the object name in e.g. inventory listings and when presenting objects that have no DESCRIPTION clause. For example

OBJECT owl ARTICLE "an"

would result in things like

```
There is an owl here.
You are carrying an owl.
```

The article is not used when mentioning the object when acting on multiple objects:

```
> take everything
(owl) Taken.
```

Note: The default article, "a" (if using english), is used for objects that have no ARTICLE declared.

For objects that should not have any article, like 'some money', an ARTICLE clause containing no statements must be used:

```
OBJECT money NAME some money
ARTICLE
```

This will lead to:

There is some money here.

instead of

There is a some money here.

Mentioned

113. mentioned = 'MENTIONED' statements

The optional MENTIONED clause gives a short form for this object given when mentioned e.g. in listings of containers or when the ALL form is used. If no MENTIONED clause is present an appropriate default message, constructed from the object name, is supplied by the system.

```
> take all
(little black book) OK!
(green pearl) OK!
```

The MENTIONED clause is also used when describing objects that have no DESCRIPTION, by inserting the article (see above) and the short description in a default message. In the following example output the article is underlined and the short description is emphasised, the rest is the default message templates.

There is a little black book, a green pearl and an owl here.

The same principle is used when constructing lists of objects in container contents lists.

Description

Objects can of course have descriptions, statements describing the object. This description will normally be printed when the player enters the location where the object currently is. It will also be given as a result of the DE-SCRIBE statement, and indirectly by executing a LOOK statement at the location where the object is. If the DESCRIPTION clause is missing the Alan system will supply a default description such as "There is a round ball here.". If there is a DESCRIPTION clause but it contains no statements the object will be 'invisible', i.e. no description of it will printed. This can be useful for objects already described by the location description, or of objects with particular properties.

Verbs

As for locations, local verbs can be declared inside an object. The verb declarations inside objects is only used when that verb is applied to the object. See *Verbs* on page 27 for details on verb declaration and usage.

3.10 Containers

```
118. container = container_header container_body container_tail
;
119. container_header = 'CONTAINER' ID
```

A container is something that can contain objects. A container can either be an object itself (in which case it is declared as an object with the CONTAINER property, see *Objects* on page 33) or be a pure container. A container that is not an object can *NOT* be manipulated directly by the player. It can, however, be manipulated indirectly, if the author has supplied some verbs to do this, such as take and drop, which usually are implemented to manipulate the inventory container. The inventory, i.e the container for everything the hero is carrying, is the most common example of an pure container. The inventory is implicitly declared, so that it exists and can be used for common purposes. It can however be redeclared if required, for example to provide limits and a different header. Its default declaration is

> CONTAINER inventory LIMITS HEADER END CONTAINER inventory.

Limits

The LIMITS clause put limitations on what and how much can be put in the container. If any of these limits are exceeded when trying to locate anything inside the container, the statements in the corresponding THEN-part will be executed and the players turn aborted. In fact these checks are performed as a consequence of the execution of a LOCATE statement (not actually the player placing anything inside the container). This means that the execution of a sequence of statements can actually be interrupted by these limitations.

The specification of an attribute, which must be a numeric default attribute, implies that the sum of this attribute of all objects in the container can not exceed the value specified. The special attribute COUNT is also allowed and indicates a limitation on the number of objects allowed.

Header and Else

HEADER is used when the contents of the container are listed. It is intended to produce something like

```
"The box contains"
```

or

```
"You are carrying"
```

The ELSE-part is used instead of the header if the container is empty.

```
CONTAINER inventory
LIMITS
weight 50 THEN "You can not lift that much."
HEADER
"You are carrying"
ELSE
"You are not carrying anything."
END CONTAINER inventory.
```

If LIMITS or HEADER is missing the Alan system supplies the default of no limits, and the messages "The \$0 contains" and "The \$0 is empty." respectively.

3.11 Events

```
133. event = event_header statements event_tail
;
134. event_header = 'EVENT' ID
;
135. event_tail = 'END' 'EVENT' optional_id '.'
```

An event is a sequence of statements executed at some location at a specified time. It can e.g. be used to create an explosion where the bomb is three moves from now or to let the ceiling of the cave fall down in five moves.

```
EVENT nearby_explosion
"Somewhere in the distance there is an explosion."
MAKE bomb gone_off.
SCHEDULE small_avalanche AFTER 2.
END EVENT.
```

Events may be scheduled and cancelled with the SCHEDULE and CANCEL statements (see *Event Statements* on page 46).

3.12 Actors

139. actor_tail = 'END' 'ACTOR' optional_id '.'

An actor is something that seems to live its own life in the game. The author refers to the actor by using the ID, and it is also the default name presented to the player.

Name

;

By means of the NAME clause, a different name can be assigned to the actor in the same way as for an object (see *Objects* on page 33).

Container property

The optional property (CONTAINER) clause may be used to indicate that this actor can be used as a container, i.e it may contain things, thereby implying that the actor is carrying the things contained. This is analogous to objects having the container property (see *Objects* on page 33).

Attributes

27. 28.	attributes = attribute '.' ! attributes attribute '.' ;
29. 30. 31. 32.	attribute = ID ! 'NOT' ID ! ID optional_minus INTEGER ! ID STRING ;

An actor can have attributes (see *Default Attributes* on page 23). These can be local attributes or override values of declared default attributes.

```
ACTOR kirk NAME Captain Kirk AT control_room
HAS health 25.
CONTAINER
HEADER "Kirk is carrying"
ELSE "Captain Kirk is not carrying anything."
DESCRIPTION
"Your superior, Captain Kirk, is in the room."
END ACTOR kirk.
```

Description

In the DESCRIPTION clause, a description of this actor can be given. The statements describing the actor will be executed when the player enters a location where the actor currently is. This description will also be given as a result of the DESCRIBE statement. An exception is if the actor is currently executing a script for which there is a separate description (see below).

Script

The SCRIPT is the actor's way of performing things. In a way it corresponds to what the hero is ordered to do by the player's typed-in commands.

Every script has a number to identify it. A script is selected by the USE statement. When a script is started it will continue until it reaches the end or another USE statement is executed for this actor.

The optional description allowed in the beginning of a script is used instead of the general description (in the beginning of the actor declaration) whenever the actor is executing that particular script. If it is not present the general description is used.

```
ACTOR george NAME George Formby

DESCRIPTION

"George Formby is here."

SCRIPT 1.

DESCRIPTION

"George Formby is here cleaning windows."

STEP

...

SCRIPT 2.

DESCRIPTION

"George Formby is tuning his ukelele."

STEP

...
```

Steps

A script is divided into steps. Each step contains statements representing what the actor will do in what corresponds to one player move. A step can be defined to be executed immediately next move, to wait a number of moves before it is executed or even to wait for a special situation (condition) to arise.

For example

```
STEP WAIT UNTIL HERO HERE
  "From the shadows a waiter emerges: $p'-Bonjour,
  monsieur', he says."
```

When an actor has executed the last step of the current script, it will do nothing more until the next USE statement is executed for this actor (the actor will be "dead", but still present at the location where it was). If this is not what is wanted, it is recommended to end each script with a new USE statement.

The Hero

There is one very special actor, the hero, which is the player. This actor is normally implicitly declared, but possible to redeclare. One situation when this is required is if you like to have attributes on the hero, such as "sleepy" or "hungry". Then a declaration like the following is possible:

```
ACTOR hero NAME me
IS NOT hungry.
VERB examine DOES
IF hero IS hungry THEN
"Examining yourself reveals a poor, hungry soul."
ELSE
"You find nothing but a poor beggar."
END IF.
END IF.
END VERB examine.
END ACTOR hero.
```

3.13 Rules

148. rule = 'WHEN' expression '=>' statements

A rule is an arbitrary expression, which, when true, results in execution of the given statements. Rules can be used to make things happen when certain situations arise, such as starting an actor when the hero enters the cave.

WHEN hero AT cave AND monster NOT active => USE SCRIPT 3 FOR monster.

The rules are tested after each actor, including the player, has made his move and after each event that is executed. Rules can be considered to be executed at the location where the last activity (actor move or event) was performed (see also *A Turn of Events* on page 59). This is important to consider especially concerning WHERE specifications (see page 48).

3.14 Start section

149. start = 'START' where '.' optional_statements

The start section defines where the player (the hero) will be at the start of the game. This must be a location. Optionally this may be followed by statements to be executed at the beginning of the game, such as hello-messages or short instructions as well as starting any actors and scheduling events.

3.15 Statements

```
150. optional_statements =
151.
                         ! statements
                          ;
152. statements = statement
        ! statements statement
153.
                ;
154. statement = output_statement
       ! special_statement
! manipulation_statement
155.
156.
               ! event_statement
157.
158.
               ! assignment_statement
159.
               ! actor_statement
160.
               ! if_statement
```

3.15.1 Output Statements

161.	output_statement	=	STRING
162.		!	'DESCRIBE' what '.'
163.		!	'SAY' expression '.'
164.		!	'LIST' what '.'

An output statement is in the simplest case just a string, i.e. any text, possibly stretching over multiple lines, surrounded by double quotes. Whenever it is executed, the string will be printed on the terminal with the following exception: if an output statement is executed at a location in the game where the hero not presently is the output will not be shown. This can be used in the following way in a script for the actor charlie_chaplin:

"Charlie Chaplin leaves the house through the front door." LOCATE charlie_chaplin AT outside_house. "Charlie Chaplin comes out from the nearest house."

If the hero is inside the house or out in the street he will now get different views of the situation.

There are some character combinations that have special meaning for the printout:

\$1 The name of the current location \$v The verb the player used (the first word) New paragraph (one empty line) \$p New line \$n Indent on a new line \$i Insert a tabulation \$t \$\$ Do not insert a space The name of the actor that is executing \$a The current object (first parameter) ŚΟ s < n > The parameter with number < n > (< n > is a digit)

Note: The a, s_0 and s_n formats must be used with care as they are not checked at compile time, e.g. you can use " s_0 " in a context where no parameter is defined which would lead to a runtime error. To avoid any run-time problems use the SAY statement with the parameter name. The use of a, s_0 and s_n formats may not be forward compatible.

Describe

The DESCRIBE statement executes the description part for an actor, an object or a location. If no such description exists a default description, such as

"There is a \$0 here."

is used instead. If the object has the container property a LIST statement is also executed for that object automatically (see below).

If a DESCRIBE statement is used for an object in the description part of a location, the system will recognise this and make sure that the object is not described more than once during the execution of a LOOK statement or when the hero enters that location. This makes it possible to use objects as parts of a location and embedding their description at the correct place in the longer description of the location.

```
"This office is dusty and probably hasn't been used for
many years."
DESCRIBE desk.
```

Say

The SAY statement will output a short description of what is referred to by the what part. If that refers to an entity (a LOCATION, OBJECT or ACTOR) it will print the name of that entity or execute its MENTIONED clause if one is available. If it refers to an attribute it will print its value (integer or string). Parameter names are also allowed in the SAY statement, which, of course will result in a short description of the entity to which it is bound, or a printing of the literal (if the parameter was a STRING or INTEGER parameter).

```
IF contents OF bottle > 0 THEN
   "In the bottle there is still"
   SAY contents OF bottle.
   "litres of water left."
ELSE
   "The bottle is empty."
END IF.
```

List

The LIST statement lists all objects in a container together with the header as specified for the container. If the container is empty the statements in the empty clause of the container is executed instead.

```
"The chest is heavy."
IF chest IS open THEN
LIST chest.
END IF.
```

3.15.2 Special Statements

```
165. special_statement = 'QUIT' '.'
166. ! 'LOOK' '.'
```

167.	! 'SAVE' '.'
168.	! 'RESTORE' '.'
169.	! 'SCORE' optional_integer '.'
170.	! 'VISITS' INTEGER '.'
171.	! 'SYSTEM' STRING '.'
	;

Quit

QUIT prints a restart question and if affirmative restarts the game, if not immediately leaves the game. Any scoring or other printouts has to be made explicitly before executing the QUIT statement.

Look

LOOK describes the current location and what it contains. The DESCRIP-TION part for the location is executed, which may include describing objects or actors by executing DESCRIBE statements. Then objects and actors that has not already been described will be described.

Save and Restore

SAVE saves the game on a file for later use with RESTORE. Both save and restore asks for a file name to use for storing and restoring.

If the player should be shown the current surroundings after a RESTORE, you will have to implement a player verb like

```
VERB oops
DOES
RESTORE.
LOOK.
END VERB oops.
```

Score

SCORE is a way of rewarding the player by giving points for certain actions. This is done by the statement

```
SCORE points.
```

for example

SCORE 25.

The first time every such statement is executed the points given are added to the players current score. SCORE without any arguments prints a message indicating the current accumulated score.

Note: The SCORE statements assumes a simple model of scoring; a number of actions is necessary to complete the game and all those are necessary to achieve the maximum number of points. For adventures having a more complex and varied scoring system (particularly if the game can be successfully finished without per-

forming all scoring actions or in multiple ways) manual scoring should instead be implemented using attributes (e.g on the player) and SET and IF statements.

Visits

The VISITS statement changes the number of times a location can be visited before the long description is presented again:

VISITS count.

The value of the argument (count) controls the number of visits to a particular location between full descriptions. The default setting (0) indicates that every time a particular location is visited its full description will be shown (which can also be expressed as: the full description will *not* be shown 0 times in between). Thus, a setting of 1 (one) would give a full description every other time the same location is visited. So

VISITS 0.

will always show long descriptions (this is also the initial setting).

Note: The familiar VERBOSE, BREIF etc. commands can be imitated using different values in the VISITS statement.

3.15.3 Manipulation Statements

Locate

The LOCATE statement is a way of transferring objects and actors. When executed, the indicated object or actor will be placed at the location given. For a description on how to specify where, see *WHERE specifications* on page 48. When an actor is located at a new location the DOES clause of that location is always executed.

One special case of the LOCATE statement is when the predefined actor HERO is located somewhere. This is analogous to what happens when the player types in a direction, i.e. the player is located at the appropriate location. Under particular circumstances, you may want to locate the player at a different location as a side effect of another action. For example:

```
EVENT explosion
   "Suddenly the door seems to bulge outwards, it bursts
   open throwing rocks and splinters everywhere. The
   impact of the explosion literally throws you back
   out in the hallway."
   LOCATE HERO AT hallway.
END EVENT explosion.
```

In this case the new location will be described and the DOES clause of that location executed.

Another special case is when locating something inside a container. The LO-CATE statement will then cause the execution of the limits of that container, and if any of the limits are exceeded the complete player turn is aborted immediately, resulting in that no more statements are executed. So if a player command should result in the location of an object inside a container, a good thing is to place the LOCATE statement as early as possible, as this enforces the limit checks in the beginning of this player turn.

Empty

The EMPTY statement locates all objects in the given container (or object or actor with the CONTAINER property) at a certain place. The meaning of the where part is as for LOCATE.

```
EMPTY inventory HERE.
"You seem to have lost most of your possessions. Well,
  you can't have everything."
LOCATE hero AT restart_point.
```

3.15.4 Event Statements

There are two statements to schedule and cancel events. The identifier must refer to an event.

Schedule

SCHEDULE event means that the given event will occur at the indicated location in the number of moves given. This can be zero moves, i.e. AFTER 0 means that the event will occur now (during this player turn). If no location is specified, HERE is assumed, i.e. it will be executed at the current location, the location where the statement itself was executed at.

The semantics of specifying a location (where) as AT id, where the identifier represents an object or an actor, is that wherever that object or actor is when the event occurs, the event will be executed at that place.

Executing a second SCHEDULE statement for the same event before it has occurred will reschedule the event to the new time.

Cancel

CANCEL will remove the event referenced from the queue of scheduled events.

```
EVENT ticking "Tick..."
```

Alan version 2.6

```
IF timer OF bomb = 0 THEN
    SCEHDULE explosion AFTER 1.
  ELSE
    DECREASE timer OF bomb.
    SCHEDULE ticking AFTER 1.
  END IF.
END EVENT ticking.
VERB defuse
  DOES
    CANCEL ticking.
    CANCEL explosion.
    "Phuuui! That was close."
END VERB defues.
START AT office.
  "The bomb is ticking..."
  SCHEDULE ticking AFTER 1.
```

3.15.5 Assignment Statements

There are a number of statements for changing values of attributes.

Make

The MAKE statement is used to set or reset boolean attributes.

MAKE door open.

Increase and Decrease

The INCREASE and DECREASE statements modifies the values of numeric attributes by increasing or decreasing them by the value of the expression given in the optional BY clause. If no BY clause is specified the attributes are changed by 1 (one).

INCREASE level OF bottle BY contents OF mug. DECREASE lives OF HERO.

Set

The SET statement is used when assigning values to numerical or string valued attributes.

```
SET mood OF king_tut TO 3.
SET hour OF clock TO hour OF clock + 1.
```

3.15.6 If Statement

The IF statement is essential for being able to vary the output and otherwise change the activities in the game. The expression is evaluated (see *Expressions* below) and if it evaluates to true, the statements following the THEN are executed, otherwise the expressions in any following ELSIF clauses are evaluated (in order) and the statements following the first expression that results in a true value is executed. If none of the expressions in the ELSIF clauses, the statements following the ELSE are executed. The ELSE clause is also optional.

```
IF minute OF clock = 59 THEN
   SET minute OF clock TO 0.
   INCREASE hour OF clock.
ELSE
   INCREASE minute OF clock.
END IF.
```

3.15.7 Actor Statements

The USE statement starts execution of a given script for a given actor. It is possible to leave out the FOR id -part when writing code within a certain actor; in this case the actor that the code is in is assumed.

USE SCRIPT 4 FOR george.

3.16 WHERE specifications

219.	where	=	'HERI	Ξ'
220.		!	'NEAI	RBY '
221.		!	'AT'	what
222.		!	'IN'	what

Many constructs in the Alan language require a specification of where the construct should operate. The general intention of a where specification is to return a location. The meaning of the constructs above are as follows

- HERE is the location where the current activity is performed. Normally this mean where the hero is, but if the expression is evaluated in an event scheduled at a particular place, that place is HERE, and the same applies to activities performed by other actors and for expressions within rules. Note that this is equivalent to AT LOCATION.
- NEARBY means any adjacent location, adjacent meaning that there exists an exit from the other location to HERE (note that the direction is from NEARBY to HERE).
- AT what means at the location of the entity referenced by the what specification (see *WHAT specifications* on page 49).
- IN what must refer a container and the expression refers to inside of that container.

Note: Not all kinds of where specifications are meaningful in all constructs requiring a where specification. An example is NEARBY which, of course, is not allowed in a LOCATE statement as this needs a definite location to locate to, and NEARBY is not specific. Instead, NEARBY is useful in IF statements to see if the monster is somewhere near.

3.17 WHAT specifications

Constructs in the grammar for the Alan language often refer to some entity defined in the Alan source. This is generally called a what specification, as it specifies what the construct refers to. The what specification may have the following forms

213.	what	=	'OBJECT'
214.		!	'LOCATION'
215.		!	'ACTOR '
216.		!	ID
		;	

OBJECT refers to the first parameter, i.e. the first object or actor referred to by the player in the input as described by the syntax. Normally this is intended for use with verbs relying on the default syntax handling; for verbs where a SYNTAX construct is specified the identifiers for the parameters should be used instead (the use of syntax declarations is strongly advised).

Note: If OBJECT is used in an expression no compile time checks can be made on class restrictions which might lead to run-time errors when referring the first parameter. The use of OBJECT in expressions might not be forward compatible.

ACTOR is always set to the actor currently active and this also applies to expressions and statements within rules as these are run once for each actor.

LOCATION is the current location, i.e. the location where the current activity is performed. This is normally the location where the hero is, but may also be

where an event is executed or where the actor currently executing (other than the hero) is.

An identifier, id, refers to the entity with that name, or a syntax parameter with that name. A syntax parameter may have the same name as an entity declared elsewhere in the source in which case the parameter overrides the entity.

Note: Not all kinds of what specifications are meaningful in all contexts. For example it is not possible to use LOCATION (nor an identifier referring to a location) as the what-part of a LOCATE statement.

3.18 Expressions

The grammar for Alan also refers to expression. This is a generic name for a number of constructs yielding a value.

```
192. expression = term
                  ! expression 'OR' term
193.
                   ;
194. term = factor
195. ! term 'AND' factor
           ;
196. factor = primary
       ! primary optional_not where
! primary binop primary
! primary optional_not relop primary
197.
198.
199.
200.
             ! primary is something
201. primary = optional_minus INTEGER
202.
               ! STRING
           ! 01...
! what
203.
               ! 'SCORE'
204.
205.
              ! aggregate where
              ! '(' expression ')'
206.
               ! attribute_reference
! 'RANDOM' INTEGER 'TO' INTEGER
207.
208.
```

3.18.1 Types of expressions

Expressions are needed e.g. in IF and SET statements. The IF statement requires a boolean expression, i.e. an expression yielding a true or false value, while the SET statement needs a numeric or a string value. Some types of expressions return a value referring to an entity (an object, an actor or a location) in the Alan source as is, for example, the case with an identifier bound to a parameter allowing actors or objects. So, the possible types of expressions in Alan are

- integers
- strings
- boolean

entities

•

3.18.2 Logical expressions

The AND and OR operators are standard binary boolean operators. AND has higher priority, but parenthesis may be used to change the order of evaluation.

IF kalif HERE AND mood OF sultan IS 0 THEN ...

3.18.3 The whereabouts of an entity

The expression primary optional_not where is used to test if a particular entity as specified by what is (or is not) at the place indicated by the where, as in

IF bottle IN inventory THEN ...

or

IF HERO NEARBY THEN ...

3.18.4 Binary and Relational operators

```
223. binop = '+'
       ! '-'
224.
           ! '*'
225.
           .
! '/'
226.
           ;
227. relop = '<>'
           ! '='
228.
229.
            ! '=='
            ! '>='
230.
           ! '<='
231.
           .
! '>'
232.
233.
            ! '<'
```

All binary operators (plus, minus, multiplication, division and the relational operators) may be used on integer expressions. A comparison between expressions may be negated by using an optional NOT.

IF temperature OF oven NOT > 100 THEN

Comparing two string expression using the binary operator '=' will make a case insensitive comparison, i.e. it will return a true value if the strings are the same without considering the case of the characters. The special identity operator, '==', only works on strings and compares the strings for an exact match (i.e. considering character case).

Two identifiers referring to entities may be compared with the '=' and '<>' operators, and may be used to test if a parameter refers to a particular entity or the same as another parameter. For example

```
SYNTAX put_in = 'put' (o) 'in' (c)
WHERE c ISA CONTAINER
ELSE "You can't put anything in the $2"
```

```
VERB put_in
CHECK o <> c
ELSE "That would be a good trick if you could do it!!"
DOES ...
```

Relational operations are not allowed on entities or strings, nor is it possible to compare values of different types.

3.18.5 The value of Attributes

The expression primary is something is used to test the setting of boolean attributes of the entity referred to by something. For example

IF bottle IS empty THEN ...

To get the value of a numeric or string attribute the expression ID 'OF' what is used.

IF s = password OF terminal THEN ...

3.18.6 Literal values

A single integer (e.g. 42) is of course a numeric expression.

The expression RANDOM integer TO integer is also a numeric value which is randomly selected between and including the two integers.

SET eyes OF first_die TO RANDOM 1 TO 6.

A string can be used in expressions and then represents a string value, e.g.

SET password OF terminal TO "xyzzy".

3.18.7 Aggregates

```
209. aggregate = 'COUNT'
210. ! 'SUM' 'OF' ID
211. ! 'MAX' 'OF' ID
;
```

Aggregates are functions to calculate values from sets of other values.

COUNT counts the number of objects at the specified place, e.g.

```
"You are carrying"
SAY COUNT IN inventory.
"things."
```

The SUM and MAX aggregates return the sum and the maximum value respectively of an attributes of all objects at the specified location. This implies that the attribute must be a default object attribute in order to ensure that the attribute is available for all objects. For example

IF SUM OF weight AT bridge > 500 THEN ...

IF MAX OF size IN inventory > size OF small_door THEN ...

The last example could be adopted to make various restrictions in the possible travels of the hero.

4 LEXICAL DEFINITIONS

4.1 Comments

Comments may be placed anywhere in the Alan source. A comment is opened by double hyphens ('--') and extends to the end of the line.

```
-- This is a comment
```

4.2 Identifiers and Names

Words used as identifiers in an Alan source may only be composed of letters, digits and underscores. The first character must be a letter.

identifier = letter (letter | digit | underscore)*

In order to be able to use reserved words as identifiers (e.g. for verbs) there is also a second kind of identifier, namely the quoted identifier.

quoted_identifier = single_quote any_character+ single_quote

A quoted identifier starts and ends with single quotes and may contain any character except quotes (including spaces). It may be used to make an identifier out of a reserved word such as LOOK. This may be useful in the definition of the verb LOOK which then would look like:

```
VERB 'look'
DOES
LOOK.
END VERB 'look'.
```

Note that normal identifiers are always translated to lower case before making any comparisons so it does not matter how you (or the player) write them (although it is easier to read if the same kind of editing is used for the same kind of words). Quoted identifiers are not changed at all, so they must always be written identically. They may also contain spaces which make them useful as long names for locations as in

```
LOCATION pluto NAME 'At the Rim of Pluto Crater' DESCRIPTION ...
```

One single quoted identifier is used as the whole name of the location so as to preserve editing and avoiding clashes with the reserved words AT and OF.

Note: Do *NOT* use a single quoted identifier as the name for anything other than locations, as the words in objects and actor names are analysed separately and assumed to be adjectives (except for the last, which is a noun). Only quote separate words to avoid clashes with reserved words. *Note:* Be careful when using quoted identifiers, especially if the player is supposed to use the word. A player can not input words containing upper case characters, underscores, spaces or other special characters or separators.

Note: To get a single quote within a quoted identifier repeat it ('Tom''s Diner').

Some of the identifiers in an Alan description is by default used as player words. This is for example the case with verb names (unless a SYNTAX statement has been declared for the VERB) and object names (unless a NAME clause has been used). If these contain special characters the player can not enter them.

4.3 Numbers

Numbers in Alan are only integers and thus may consist only of digits.

number = digit+

4.4 Strings

The string is the main lexical component in an Alan source. This is how you describe the surroundings and events to the player. Strings, therefore, are easy to enter and consist simply of a pair of double quotes surrounding any number of characters. The text may include newline characters and thus may cover multiple lines in the source.

string = '"' any_character+ '"'

When processed by the Alan compiler, any multiple spaces, newlines and tabs will be compressed to one single space as the formatting to fit the screen is done automatically during execution of the game (except for embedded formatting information, as specified in *Output Statements* on page 42). You may therefore write your strings any way you like, they will always be neatly formatted on the players screen.

Note: As strings may contain any character a missing double quote may lead to many seemingly strange error messages. If the compiler points to the first word after a double quote and indicates that it has deleted a lot if IDs (identifiers), this is probably due to a missing end quote in the previous string.

Note: To get a double quote within strings repeat it ("The sailor said ""Hello!"".").

4.5 Files

It is possible to write one adventure using many files, having different parts in different files, thus giving an opportunity for some rudimentary kind of modularisation. The method for this is the <code>\$include</code> construct.

```
include = '$INCLUDE' quoted_identifier
```

where the quoted identifier is the name of the file to include. The *\$include* may be placed anywhere in a file and the effect will be the same as if the contents of the named file had been inserted at that position in the file. Includes may be nested.

5 EXECUTION OF AN ADVENTURE

5.1 A Turn of Events

The player in a way controls the execution of an Alan adventure. Each of his inputs are taken care of and acted upon by the run-time system. The execution of an Alan adventure starts by executing the start section. Then the player is prompted for a command.

The player input is analysed according to the explicit and implicit syntax rules and converted to an execution of verb bodies (global and in possible parameters) or exits (in case of directional commands).

After the players command has been taken care of all rules are evaluated and possibly executed. Then each of the other actors execute one step (if active) and for each actor the rules are evaluated again. Finally any events that are scheduled are fired before prompting the player again.

So to summarise:

```
get and execute a player command
evaluate all rules
for each actor
    execute one step (if active)
    evaluate all rules as above
end
check for and execute any pending events
```

Then the user is prompted for another command and everything is repeated.

A player command may be either a verb or a direction. A verb is executed by checking the syntax of the input, performing any preconditions (checks) and then executing the verb bodies (as described in *Verbs* on page 27). A directional command is executed by finding any exit in that direction, evaluating the checks and the body (if any) of that exit and locating the *hero* at the new location.

5.2 Player Input

The syntax defined in the Alan source is the basis for what the player is allowed to input. Commands with these formats form the basic statements available to the player. In addition the following combinations and variations are possible:

- concatenating of statements using AND or THEN, like
 > open the door then enter
- the use of IT to refer to the last object mentioned in the previous command, e.g
 - > take the book and read it

- references to multiple objects using AND, this allows
 > take the blue vase and the pillow
- reference to multiple objects using ALL or EVERYTHING
 > drop all
- excluding objects using BUT or EXCEPT, like:
 > wear everything except the bowler hat
- the use of THEM to refer to the multiple objects referenced in the last command, e.g.
 - > remove the hat and the scarf then drop them

The reference to multiple objects (or actors) in a position is, of course, only allowed if the adventure author has allowed it by using a multiple indicator in the syntax definition (see *Syntax Definitions* on page 25). These variations are built in and handled automatically by the run-time system.

The interpreter also automatically restricts parameter references to objects and actors at the current location. I.e. the player can only refer to objects and actors that are present in his input. For hints on how to allow references to objects and actors that are not at the current location, refer to *Imaginary Objects* on page 67.

Another restriction placed on the player input by the interpreter is that the words the player is allowed to use can only contain alphabetic characters. This must be kept in mind when naming verbs that use the default syntax (an explicit SYNTAX statement can always specify other player words to trigger the verb).

5.3 **Run-time Contexts**

When the player enters a command the Alan run-time system evaluates the various constructs from the adventure description (source) as described above. Depending on the players command evaluation of different parts of the adventure may be triggered. These parts all have different conditions under which they are evaluated and also have different contexts. Four different execution contexts can be identified:

- execution of a verb, during the execution of a verb (the syntax and verb checks and the verb bodies), which is the result of the player entering a command that was not a directional command, parameters are defined and may be referenced in the statements and expressions. Also the ACTOR is set to the hero and LOCATION to the location where the hero is (HERE refers to the location of the hero).
- execution of descriptions, these are started as the response to a directional command, a LOOK or DESCRIBE statement, or a LO-CATE statement operating on the hero. During this no parameters are defined, ACTOR is set as above, and LOCATION of course to

the current (or new) location. The description clauses for objects and locations as well as the DOES part of locations are evaluated in this context. DOES-parts are executed for all actors entering a location with ACTOR set to the current actor.

- execution of actors and rules, each actor performs his step and after each actor all rules are executed. In these contexts no parameters are defined but ACTOR is set to the actor that is executing or was executing immediately preceding the rules. So you could say that rules are run for each actor, and LOCATION is set to that of the executing actor (HERE refers to where the executing actor is).
- execution of events, no parameters and no actor is defined. The location is set to where the event was scheduled to be executed (see also *Event Statements* on page 46).

So the execution of various parts of the adventure source can also be said to have a number of different focuses, meaning where the action is considered to take place:

- the hero the actions of the player are always focused on the hero and the actions performed are always related to where the hero is
- an actor steps executed by an actor are always focused where the actor is
- an event code executed in events are focused where the event was specified to take place (see *Event Statements* on page 46).
- a rule rules are executed once after each actor (including the hero) with the focus set to where that actor is

5.4 Moving Actors

The main way to move actors are the exits (see *Locations* on page 31). They, of course, only apply to the hero, but are executed if the player inputs a directional command, i.e. a word defined as the name for an exit in any location. First the current location is investigated for an exit in the indicated direction, if there is none an error message is output. Otherwise that exit is examined for CHECKs which are run according to normal rules (see *Verbs* on page 27). If no CHECK was present or if the check passed the statements in the body (the DOES-part) is executed. The hero is then located at the location indicated in the exit header, which will result in the description of the location (by executing the DESCRIPTION-clause of the location) and any objects or actors present (by executing their DESCRIPTIONs).

When any actor (including the hero) is located at a location, the DOES-clause of that location is executed as if the actor had moved into that LOCATION. The actor which was moved will be the ACTOR even though the movement was not caused by himself (but the result of an event, for example). So this is

also the last step in the sequence of events caused by locating the hero somewhere.

6 HINTS AND TIPS

This chapter will give you some ideas about how the various features of Alan may be used to implement common features in an Adventure game. These are only suggestions and you are, of course, welcome to invent your own, but these are probably some ideas that can get you started.

6.1 Use of Attributes

Attributes are primarily used for holding status information about the object, actor or location to which it belongs. This allows, for example, the water bottle to contain three levels of water.

```
OBJECT bottle

HAS level 3.

VERB drink

DOES

IF level OF bottle > 0 THEN

DECREASE level OF bottle.

ELSE

"There is no more water in the bottle."

END IF.

END VERB drink.

END OBJECT bottle.
```

Another example is the broken mirror.

```
OBJECT mirror
IS NOT broken.
VERB break
DOES
MAKE mirror broken.
END VERB break.
END OBJECT mirror.
```

The appropriate verbs defined in the objects may then modify the attributes and thus update the status information.

But attributes defined for all objects also allows a kind of classification of the objects (or locations or actors as appropriate). If the following declaration is made

```
OBJECT ATTRIBUTES NOT takeable.
```

all objects receive the attribute "takeable" and if the attribute is not specifically redeclared for an object it will not be takeable. Note however that the semantic meaning of "takeable" must be implemented e.g. in the verb "take":

```
VERB take
CHECK OBJECT IS takeable
ELSE "You can't take the $o."
DOES
LOCATE OBJECT IN inventory.
END VERB take.
```

In the same way restrictions concerning what is possible to eat, drink, open etc. may be implemented. This use of attributes to classify objects is "actionoriented", i.e. they imply that a particular action (verb) is applicable to the object.

An alternate approach is to classify objects after their characteristics. Consider:

```
VERB take
CHECK OBJECT IS NOT heavy
ELSE "That is much too heavy."
AND OBJECT IS NOT animal
ELSE "The $0 moves quickly away, just far enough
for you not to reach it."
DOES
LOCATE OBJECT IN inventory.
END VERB take.
```

This approach is more "class-oriented" as the objects are classified and a verb is possible to apply to certain classes of objects and not to others. This approach is more elegant but is harder to keep track of as you introduce new objects (which class or even classes does a new object belong to?).

6.2 **Descriptions**

The attributes are also used when presenting information about status to the player. The attributes are tested in IF-statements to modify the DESCRIP-TIONS and possibly even the short description in the MENTIONED sections. For example:

```
OBJECT mirror
IS NOT broken.
DESCRIPTION
"On the wall there is a beautiful mirror with an
elaborate golden frame."
IF mirror IS broken THEN
"Some moron has broken the glass in it."
END IF.
VERB break
DOES
MAKE mirror broken.
END VERB break.
END OBJECT mirror.
```

To use this feature with the short descriptions makes the adventure feel a bit more consistent.

```
OBJECT bottle

HAS level 3.

ARTICLE ""

MENTIONED

IF level OF bottle > 0 THEN

"a bottle of water"

ELSE

"an empty bottle"

END IF.

END OBJECT bottle.
```

> inventory
You are carrying
 an empty bottle

6.3 Common Verbs

As your library of adventures grow you will find that some verbs are always needed, and always function the same way. Examples are "take", "drop", "invent", "look", "quit" and so on. It is advised to use an include file contain these verbs as well as their syntax definitions and any synonyms. Attributes needed for these particular verbs could also be placed in a default attribute declaration in this file.

All your adventures may then include this file, making these features immediately accessible when you start a new adventure. All that this takes is some thought as to what names to use for the open, openable, takeable etc. attributes as discussed in *Use of Attributes* on page 63.

6.4 Doors

Another common feature is the closed door. Here's how to implement it.

```
OBJECT treasury_door AT hallway
  VERB open
    DOES
       MAKE treasury_door open.
       MAKE hallway_door open.
  END VERB open.
END OBJECT treasury_door.
LOCATION hallway
  EXIT east TO treasury
    CHECK treasury_door IS open
      ELSE "The door to the treasury is closed."
  END EXIT.
END LOCATION hallway.
OBJECT hallway_door AT treasury
  VERB open
    DOES
       MAKE treasury_door open.
       MAKE hallway_door open.
  END VERB open.
END OBJECT treasury_door.
LOCATION treasury
  EXIT west TO hallway
    CHECK hallway_door IS open
      ELSE "The door to the hallway is closed."
  END EXIT.
END LOCATION treasury.
```

Note that we need two doors, one at each location, but they are synchronised by always making them both open or closed at the same time. The check in the EXITS makes sure that the hero can not pass through a closed door.

6.5 Containers

Containers are either pure containers or objects or actors with the container property. A pure container is always considered to be where the hero is. This means that the inventory (what the hero is carrying), his clothes etc. are suitable to be pure containers.

For a container to be directly manipulable by the player it must be an object (or actor). This means that it always is located at a particular location in the same way as other objects. A container (in the following the term container is used to refer to objects with the container property) is always open, i.e the objects it contain are always accessible.

To be able to "close" a container, i.e. to make it impossible for the hero to take or see things inside a container, the following technique may be used (other techniques may be possible and even better!). Create an extra object with the container property, this container is used as a temporary storage for objects in the first container (the one the player is seeing). Place this at a location not accessible to the player (the limbo location Nowhere always comes in handy!).

The verbs "open" and "close" then get the following definition within the object:

```
OBJECT chest AT treasury
  CONTAINER
  IS NOT open.
  VERB close
    DOES
       MAKE chest NOT open.
       EMPTY chest IN chest contents.
  END VERB close.
  VERB open
    DOES
       MAKE chest open.
       EMPTY chest_contents IN chest.
       "Opening the chest reveals its contents."
       LIST chest.
  END VERB open.
END OBJECT chest.
```

The trick used here is to make all the things in the container disappear when it is closed. To do this, the extra container chest_contents is used as a temporary holding place for the things inside the chest. Note that we need to make chest_contents an actual object since pure containers are always accessible (they are where the hero is!). When the chest is opened again we simply empty the contents of the chest_contents container into the chest, and *Voila*!

6.6 Imaginary Objects

Sometimes you need to make it possible for the player to refer to things either far away, things that are not really objects or that may be at many places at once. Examples of these are a distant mountain that may be examined through a set of binoculars, the melody in "whistle the melody", and water or walls. The problem here is that the Arun interpreter assumes that all objects referenced by the player must be present (the cause for this is two-fold: first, this is the most common case which would otherwise be necessary to handle manually in checks for every verb and secondly, it is used to solve ambiguities between different objects with the same name).

For objects that should be visible from a distance the easiest method is to introduce a 'shadow object'. This is a second object acting on behalf of, or representing, the distant object at the locations where it should be possible to refer to it. For example:

LOCATION hills : END LOCATION hills. OBJECT mountain AT hills : END OBJECT mountain. LOCATION scenic_vista NAME Scenic Vista END LOCATION scenic_vista. OBJECT shadow_mountain NAME distant mountain AT scenic_vista DESCRIPTION "Far in the distance you can see the Pebbly Mountain raising towards the sky." END OBJECT shadow_moutain.

This would allow for example at scenic_vista:

Scenic Vista.
Far in the distance you can see the Pebbly Mountain raising
towards the sky.
> look at mountain through the binoculars

which would otherwise be impossible. If the mountain should be visible and manipulable from a number of locations, you might implement one shadow object for each location but this is a bit tedious if they are identical. One trick here is to use something like the following rule:

WHEN hero AT scenic_vista OR hero AT hill_road => LOCATE shadow_mountain AT hero.

This will ensure that whenever the hero moves to any of the places from where the mountain is visible, the shadow_mountain is sure to follow.-However, as the rules are executed *after* the hero has moved, a better strategy might be to make the shadow_mountain 'silent', i.e. to have no description. Instead the description of it should be embedded in the description of the adjacent locations. Yet another possibility would be to move the pseudo-object around using statements in the exits, like

```
LOCATION scenic_vista NAME Scenic Vista
EXIT east TO hills
DOES
LOCATE shadow_mountain AT hills.
END EXIT east.
END LOCATION scenic_vista.
```

Objects that are always present, such as the air or the parts of the heroes body, may be treated like normal objects. I.e. they are defined as the objects they represent. They are then placed in a container that is not an object, which makes the objects always accessible, since containers (that are not objects) are considered to be where the hero is (cf. the inventory). This is also a simple way to create other compartments on the hero, such as a belt.

```
CONTAINER belt
  LIMIT count 2
    ELSE "You can't fit more in your belt."
END CONTAINER belt.
VERB invent
  DOES
    LIST inventory.
    LIST belt.
END VERB invent.
CONTAINER pseudo
END CONTAINER pseudo.
OBJECT air IN pseudo
  VERB breathe
    :
  END VERB breathe.
END OBJECT air.
```

6.7 Actors

Actors are a vital component to make a story dynamic. They move around and act according to their scripts. To make the player aware of the other actors actions they need to be described. This must be done so that the player always get the correct perspective on the actions of the actors.

A way to ensure this is to rely on the fact that output statements are not shown unless the hero is at the location where the output is taking place. This means that for every actor action, especially movement, you need to first describe the actions, then let the actor perform them and, finally, possibly describe the effects.

An example is the movement of an actor from one location to another. In this case the step could look something like

"Charlie Chaplin goes down the stairs to the hallway." LOCATE charlie_chaplin AT hallway. "Charlie Chaplin comes down the stairs and leaves the house through the front door." LOCATE charlie_chaplin AT outside_house. "Charlie Chaplin comes out from the nearest house."

An actor is described, for example, when a location is entered or as the result of a LOOK in the same way as objects are. This means that a good idea is to include the description of an actors activities in the description of him. One way to do this would be to use attributes to keep track of the actors state and test these in the description clause.

```
ACTOR george NAME George Formby
IS
NOT cleaning_windows.
NOT tuning.
DESCRIPTION
IF george IS cleaning_windows THEN
"George Formby is here cleaning windows."
ELSIF george IS tuning THEN
"George Formby is tuning his ukelele."
ELSE
"George Formby is here."
END IF.
```

Although quite feasible, this is a bit tedious. As, at least a part of, the state is indicated by the script the actor is executing, this could be used to avoid the potentially large IF-chain. The optional descriptions tied to each script will be executed instead of the main description when the actor is following that script. So this would allow

```
ACTOR george NAME George Formby

DESCRIPTION

"George Formby is here."

SCRIPT 1.

DESCRIPTION

"George Formby is here cleaning windows."

STEP

...

SCRIPT 2.

DESCRIPTION

"George Formby is tuning his ukelele."

STEP

...
```

This makes it easier to keep track of what an actor is doing. Another hint here is to describe the change in an actors activities at the same time as executing the USE statement, like

```
EVENT start_cleaning
  USE SCRIPT 1 FOR george.
  "All of a sudden, George starts to clean the windows."
END EVENT.
```

This makes the descriptions of changes to be shown when it takes place and the description of the actor is always consistent. You can, of course, still have attributes describing the actors state to customize the description of the actor

Alan version 2.6

on an even more detailed level, but it generally suffices to describe an actor in terms of what script he is executing.

6.8 Distant Events

A slight problem with the feature that output is not visible unless the hero is present, is that a description of an event might not always be presented to the player.

```
EVENT explosion
 "A gigantic explosion fills the whole room with smoke
 and dust. Your ears ring from the loud noise. After a
 while cracks starts to show in the ceiling, widening
 fast, stones and debris falling in increasing size
 and numbers until finally the complete roof falls down
 from the heavy explosion."
 MAKE LOCATION destroyed.
END EVENT.
```

If the hero isn't at the location where the event is executed, he will never know anything about what has happened. The solution is to create an event that goes of where the hero is.

```
EVENT distant_explosion
   "Somewhere far away you can hear an explosion."
END EVENT.
...
IF HERO NEARBY THEN
   SCHEDULE distant_explosion AT HERO AFTER 0.
...
```

6.9 Questions and Answers

Sometimes it may be necessary to ask the player for an answer to some question. One example is if you want to confirm an action. The following example delineates one simple way to do this which could be adopted for various circumstances.

```
ACTOR hero IS NOT quitting.

END ACTOR hero.

SYNTAX

'quit' = 'quit'.

yes = yes.

SYNONYMS

y = yes.

q = 'quit'.

VERB 'quit' DOES "Do you really want to give up?

Type 'yes' to quit, or to carry on

type your next command."

MAKE hero quitting.

SCHEDULE unquit AFTER 1.

END VERB 'quit'.
```

```
VERB yes CHECK hero IS quitting
ELSE "That does not seem to answer any question."
DOES QUIT.
END VERB yes.
EVENT unquit MAKE hero NOT quitting.
END EVENT unquit.
```

Thanks to Tony O'Hagan (aoh@maths.nott.ac.uk) for this excellent idea.

6.10 Structure

A good thing to do when designing an interactive fiction story is to separate the geography from the story. In Alan you can use the include facility to structure your Alan source. An method could be to place the description of each location in a separate file together with any objects that could be considered part of the scenery or at least is not only a tool in a puzzle. These files can then be included in a 'map' file which in turn is included by the top level file.

The story line can be divided into files too, one for each 'scene'. A scene being comments describing the important things that are suppose to happen, any prerequisites and objects, events, rules etc. which are specific for this part of the story.

This strategy will both give you a better structure of your adventure as well as leading you to designing a better story, much like the storyboarding technique used in making movies or plays.

6.11 Debugging

To simplify the development of adventures written in the Alan language, the interpreter incorporates some features for debugging. There are a few debugging switches available when starting the interpreter:

```
-t Enable trace mode
-s Enable single instruction trace
-d Enable debug mode
```

Trace mode will print out information about each invocation of the instruction interpreter, making it easy to see which parts of the code are being executed.

Single instruction trace will, in addition to trace mode information, also trace every single Acode instruction.

Finally, debug mode will execute the start up sequence and then prompt for a debug command with

ABUG>

Note: None of the above switches are effective unless the adventure was compiled with the debug option set (see *Options* on page 21).

Abug may also be entered by typing the single command

> debug

during the execution of an Adventure that was compiled with the debug option.

A question mark or an 'h' will give a brief listing of the commands available in Abug:

Display a list of all actors. a Display a list of all containers. С Display a list of all events and their status. е Go on. I.e proceed by executing the next turn. Abug g will stop and prompt for a new command again before the player is next in turn. Display a list of all locations. 1 Display a list of all objects. ο Quit the adventure (and Abug). q s Toggle single instruction trace. Toggle trace mode (off and on). t Exit Abug, i.e proceed without stopping. х

The commands A, C, L and O may optionally be followed by a number. Abug will then display detailed information about the entity requested, such as values of attributes, its present location etc.

Currently there is no way to modify anything using Abug.

The following is a short excerpt from a debugging session (user input in bold face):

```
<Arun, Adventure Interpreter version 2.6 alpha>
<Version of 'saviour' is 2.6(0)a>
Welcome to the game of SAVIOUR!
[introductory text deleted for brevity]
ABUG> s
Step on.
ABUG> t
Trace on.
ABUG> g
> n
<EXIT 1 (n) from 22 (Outside The Tall Building),
Executing:>
dd9: PUSH
                1
dda: SCORE
                1
                          (5)
ddb: RETURN
      _____
```

<EXIT 1 (n) from 22 (Outside The Tall Building), Moving:>

 de4:
 PUSH
 4

 de5:
 PUSH
 6229

 de6:
 PRINT
 6229,

 de7:
 RETURN

 4 "Hall" _____ de8: PUSH158de9: PUSH6235dea: PRINT6235,158"Inside the entrance is a hallway full of dust and pieces of the ceiling has fallen to the floor. At the west end is a staircase, and to the south is the exit." 1 deb: PUSH dec: DESCRIBE 620: PUSH 30 621: PUSH 1428 622: PRINT 1428, 30 " To the east is a folding door." 623: PUSH 6 624: PUSH 624: PUSH 1 625: ATTRIBUTE 1, б (1) 626: IF TRUE 13 1446 144 627: PUSH 628: PUSH 629: PRINT 1446, 13 " It is closed." 62a: ELSE 62f: RETURN ded: RETURN _____ ABUG> a ACTORS: 17: Hero ABUG> **a 1**7 ACTOR 17 : Hero Location = 23 Hall Script = 0Step = 0Attributes = ABUG> o OBJECTS: 1: door 2: rats 3: spool of computer tape 4: old book 5: 3 metre long ladder 6: rather heavy computer terminal 7: small coin 8: birds nest 9: set of rusty keys 10: clock 11: drawer 12: desk

73

```
13: dirty manual
     14: computer
     15: vending machine
     16: old mouldy candy bar
ABUG> o 6
OBJECT 6 : rather heavy computer terminal
    Location = 30 Terminal Room
    Attributes =
      1: 1 (takeable)
      2: 1 (readable)
      3: 0 (openable)
      4: 0 (startable)
      5: 1 (examinable)
      6: 0 (connected)
      7: 0 (showing_msg1)
      8: 0 (showing_msg2)
ABUG> q
```

Lines of '+' characters indicates the start of interpretation, thus they can be present inside other single step traces (like the DESCRIBE in the example above). Likewise lines of '-' indicates the return from one such level of interpretation.

7 ADVENTURE CONSTRUCTION

This chapter will give a few clues on how to be a successful adventure author, because creating a *good* adventure is more like writing a book than writing a program (although Alan can be viewed as a kind of programming language).

7.1 Getting an Idea

As with a book, the success or failure depends on how intriguing the story is, how hooked you can get the reader (in our case the player). So, the first step *must* be to get a good idea. This may be hard or easy but with time you, like a good author, learn to pick up ideas when you get them in ordinary every-day life, and store them for later use.

A seemingly simple idea might also be developed into a good adventure if it is placed in the correct setting and supplied with additional features, tricks and problems.

When you have a good idea, try to refrain from typing it in directly in a text editor and compile it with Alan. Instead, write the story down as if it were the story line for a book or a movie. Where appropriate, insert hints on various diversions and alternate paths that come to mind, but try to stay mainly with the main story from beginning to the preferred end. Then, let a close friend read it.

7.2 Elaborating the Story

After having rewritten the story line once or twice, start creating the scenery. If your setting is small, you could draw a map of the locations needed, but a better way is probably to make a list of major locations first (those essential to the story). For each location note what important properties the location must have and which objects are necessary (just as notes, *don't* create the Alan declarations yet!). For each object, make a small note on what the object is needed for (by the player!).

This may also be done using a scene by scene approach. By this we mean that the story is segmented into scenes (and maybe also acts) like in a play. For each act and scene you do the above. This makes it easier to get an overview over a larger adventure.

I also suggest that you also create a story on a level above the actual game, at least in your own mind. This story should explain why the game-world exists and thus give a consistency to the text that you will present to the player. Nobody likes an adventure without a cause. This story or world of ideas need not be revealed to the player.

This also applies to the narrator, i.e. the imaginary person or creature that carries out the conversation with the player. Create an image of him or it and stick to it. Receiving comments about your (limited) progress in the game might be funny as long as they are not out of character.

7.3 Implementing it

At last it is time to sit down at the terminal. Divide the adventure text into files containing global verbs, the map (possibly divided further according to the scenes), the actors (perhaps one file for each actor) and a main file including the other files. This makes it easy to handle the adventure and you may also ask your friend to participate in the development by giving him a few files to work on.

First, just declare the locations and connect them with exits. Do not work on the "purple prose" descriptions yet. The Alan system supplies good defaults for descriptions and so on, so use these while developing the structure of the adventure. Do not bother even with the details of making it impossible to pick up the elephant, etc.

Play the adventure continuously during the development, but do not try the things you plan to make impossible later. Just go through it according to the line you planned the story to follow. A hint here is to use a separate file for the start section. In this file you can easily set up the situation you wish to test while not having to tire yourself by playing the adventure from the start every time.

7.4 Polishing the Adventure

So, now you have a working adventure, a bit bare bones, but still the story plays the way you planned. Now it is time to insert all the nice descriptions, the limitations and perhaps the extra things to divert and hinder the hero. Just be careful not to fall into the locked-door-syndrome. Too many adventures have been tedious to play because you need to find-key/get-key/unlock-door-with-key/open-door (anyway, why do people go around locking doors and throwing away the keys). Think big.

Start by fixing the verbs so that they prohibit the impossible. Introduce as many synonyms as you can think of, this makes the adventure so much more playable.

Create the location descriptions. Remember to use the same style in all your descriptions; breaking out of style does not look good in the eyes of the adventurous. The descriptions must give the player the correct image, the brain is still the best graphic interface available, but they should also plant ideas in the player on how to solve the problems you place before him.

Another thing to aim for is the feeling which a player gets when he somehow finds information explaining things he has encountered earlier in the game. Here, as always, it is good advice to ask a friend to read the texts and convey his or her impressions (remember you know it all because you wrote it!).

Lastly fill in the adjectives for the objects, their descriptions and short descriptions (if needed).

Now, at last, your adventure game is ready to meet its audience.

A **RUN-TIME MESSAGES**

This appendix describes the errors that may occur during the running of the adventure, i.e during interpretation of the generated Acode. There are two classes of errors, system errors that are fatal and abort the adventure, and user errors which are not fatal but abort the execution of the current player command and discard the rest of the user input, which is a normal part of the interaction between the player and the Alan run-time system.

A.1 System Errors

System errors are errors caused by internal malfunctions. Mainly these are implementation errors (aka. bugs!), but may also result from user errors. The system error messages also have a purple prose style to fit in with your game:

As you enter the twilight zone of Adventures, you stumble and fall to your knees. In front of you, you can vaguely see the outlines of an Adventure that never was. SYSTEM ERROR: Can't open adventure code file.

Author Errors

The following system errors are in some sense caused by the Adventure author (you).

Out of memory.

The adventure was so large that the interpreter could not allocate enough dynamic memory for it. Try to finish other running applications (does not work or is not possible on all systems), get more real memory, or complain to the Alan implementors (see appendix G, *FU-TURE DEVELOPMENTS*, on page 109 for how to reach us).

Incompatible version of ACODE program.

The interpreter you are using have a different version than the Alan compiler used to compile the adventure. Use a different Arun or recompile the adventure with the matching compiler.

Note: the Arun switch '-d' will, beside entering debug mode, also print the version of both the Arun interpreter and the version of the Alan compiler used to compile the adventure.

Recursive LOOK.

This message is shown when a LOOK statement is executed as a result of a LOOK, i.e. a recursive LOOK! The LOOK statement should only be used in global verb bodies, *not* in descriptions of LOCATIONS and OBJECTS as there is a definite risk that it will be executed as the effect of a LOOK, either explicit or implicit (by the hero entering that location!). Locating something inside itself.

This means that an attempt to locate an object (that is a container) inside itself has been made. This might happen if the adventure author has neglected to check this in a verb like put_in = 'put' (o) 'in' (c).

Non-existing parameter referenced.

A parameter that wasn't available was referenced. This is probably due to using for example \$2 in a string in a context where the syntax was restricted to only one parameter. This may avoided by using the SAY statement instead of the embedded string parameter references, which would result in compile time checking avoiding the risk of having this happen to the player.

Note: Parameter references embedded in strings are *not* currently checked during compile time.

Note: Embedded string references (\$1, \$0, etc.) is *not* guaranteed to be forward compatible (i.e. it may be removed in future versions).

Player Errors

Can't open adventure code file.

```
Can't open adventure data file.
```

The player attempted to run an adventure for which there were no code or data file available, probably a misspelling.

Could not read all ACD code.

Checksum error in .ACD file (%1 instead of %2).

These two messages indicate problems in the adventure files. Possibly caused by transfer problems of the **.acd** and **.dat** files which must be made in binary mode.

Implementor Errors

Any other text in a system error message is really a SYSTEM ERROR. Scribble down the text and contact the implementors (see appendix G, *FUTURE DEVELOPMENTS*, on page 109). If possible supply the source for your adventure, a trace of the few last player commands (if possible with single step and trace turned on, see *Debugging* on page 71)

A.2 Input Responses

Various messages are printed for the benefit of the player. Most messages probably come from the adventure itself, but there are a small set of messages

given directly by the Arun interpreter. They are presented below using the Alan STRING-format, i.e. containing the special character combinations described in *Output Statements* on page 42. These standard messages exist for all languages and are selected depending upon the setting of the language option. See also *Message Identifiers* on page 81 for a complete list of all messages and their identifiers.

Huh?

No input at all.

I don't understand.

The input did not follow any syntax the Arun parser knows about.

I don't know what you mean by 'all'.

The player input ALL, but the Arun parser could not find any objects or actors that it could refer to.

I don't know what you mean by 'it'.

IT may only be used when the previous command contained a reference to one object or actor.

I don't know what you mean by 'them'.

THEM refers to the set of objects or actors mentioned in the previous command. If there were no multiple parameters in the previous player command, Arun will issue this message.

You can't refer to multiple objects with \$v.

The syntax for the indicated verb did not allow multiple parameters.

I can't guess what you want to \$v.

The verb required a parameter.

You must supply a noun.

The player started to specify an object or actor but only supplied the adjectives.

You must give an object after 'but'.

In a command containing ALL BUT, the player must also give the object or objects excluded.

You can only use 'but' after 'all'.

The words BUT and EXCEPT may only be used after ALL.

That doesn't leave much to \$v!

The player used an ALL BUT construct which explicitly excluded everything matched by the ALL.

```
I don't know which $1 you mean.
```

There were multiple objects (or actors) that matched the description given by the player. More adjectives are necessary to distinguish between them.

```
I can't see any $1 here.
```

The player referred to an object or actor that was not present.

You can't go that way.

A directional word was used but there is no exit in that direction.

You can't do that.

Somehow Arun found no verb body to execute. This may be a situation overlooked by the author or the player may be trying to do something that is not possible.

```
You can't $v the $1.
```

A variation of the above.

<More>

The classic message when the screen is full. The player should press RETURN to proceed.

Sorry, save failed.

When executing a SAVE, the file system indicated some error.

Sorry, could not open the save file.

When executing a RESTORE, Arun found no save file (file called **<ad-venture>.sav**) on the current directory.

- Sorry, the save file was created by a different version. The save file found was created by a different version of the Alan compiler.
- Sorry, the save file did not contain a save for this adventure. The indicated save file did not contain a save of this adventure.

A.3 Message Identifiers

The following messages are available, and may be modified using the MES-SAGE statements (see section 3.5 on page 25). All messages are available in all supported languages but below only english message texts are shown.

A description of the usage of some of the messages can be found in the section *Input Responses* above.

```
HUH,
      "Huh?",
WHAT,
      "I don't understand.",
WHATALL,
      "I don't know what you mean by 'all'.",
WHATIT,
      "I don't know what you mean by 'it'.",
WHATTHEM,
      "I don't know what you mean by 'them'.",
MULTIPLE,
      "You can't refer to multiple objects with '$v'.",
WANT,
      "I can't guess what you want to $v.",
NOUN,
      "You must supply a noun.",
AFTERBUT,
      "You must give an object after 'but'.",
BUTALL,
"You can only use 'but' after 'all'.",
NOTMUCH,
      "That doesn't leave much to $v!",
WHICHONE,
      "I don't know which $1 you mean.",
NOSUCH,
"I can't see any $1 here.",
NOWAY,
      "You can't go that way.",
```

```
CANTO,
"You can't do that.",
CANT,
      "You can't $v the $1.",
NOTHING,
      "There is nothing here that you can $v.",
SEEOBJECT1,
      "There is",
SEEOBJECTCOMMA,
     "$$, ",
SEEOBJECTAND,
      "and ",
SEEOBJECTEND,
      "here.",
SEEACTOR,
     "is here.",
CONTAINS1,
      "The",
CONTAINS,
      "contains",
CONTAINSCOMMA,
      ", ",
CONTAINSAND, "and ",
CONTAINSEND,
      "$$.",
EMPTY1,
"The",
ISEMPTY,
      "is empty.",
HAVESCORED,
      "You have scored",
SCOREOUTOF,
      "points out of",
```

```
UNKNOWNWORD,
      "I don't know that word.",
MORE,
      "<More>",
AGAIN,
"(again)",
SAVEWHERE,
      "Enter file name to save in",
SAVEOVERWRITE,
      "That file already exists, overwrite (y) ? ",
SAVEFAILED,
      "Sorry, save failed.",
SAVEMISSING,
      "Sorry, could not open the save file.",
SAVEVERSION,
      "Sorry, the save file was created by a different version.",
SAVENAME,
      "Sorry, the save file did not contain a save for this
      adventure.",
RESTOREFROM,
      "Enter file name to restore from",
RETRY, "Do you want to restart (y) ? ",
DEFARTICLE,
      "a",
```

B ALAN LANGUAGE GRAMMAR

The Alan language is defined formally below using a BNF-form. This is a set of rules defining exactly what constructs are legal in an Alan source. The rules are numbered for easy reference.

The BNF form divides the structure of the input source into smaller parts (rules) which in turn are defined by other rules. For example rule 1 says that an ADVENTURE (in this case an Alan program) consists of options, defaults, units and a start section. In rule 149 we see that the start section is in turn defined to contain the reserved word START followed by a reference to where to start, followed by a period and some optional statements.

The equal sign (=) may be read "consists of" or "is defined as". The exclamation mark indicates a choice between the two different constructs, for example in rule 6 through 8 one can see that an option may either be a single identifier, an identifier followed by another identifier or an identifier followed by an integer. The semicolon indicates the end of the definition of the symbol on the left hand side of the equal sign.

```
1. adventure = optional_options units start
              ;
4. options = option
   ! options option
;
 5.
 6. option = ID '.'
 7. ! ID ID '.'
8. ! ID INTEGER '.'
           ;
 9. units = unit
10. ! units unit ;
11. unit = object_default
     ! location_default
! actor_default
12.
13.
       ! actor_den
! messages
! rule
! synonyms
! syntax
! verb
! location
14.
15.
16.
17.
18.
19.
20.
        ! object
21.
        ! container
22.
         ! actor
23.
         ! event
24. location_default = 'LOCATION' 'ATTRIBUTES' attributes
                     ;
25. object_default = 'OBJECT' 'ATTRIBUTES' attributes
                  ;
26. actor_default = 'ACTOR' 'ATTRIBUTES' attributes
27. attributes = attribute '.'
28. ! attributes attribute '.'
29. attribute = ID
```

30. ! 'NOT' ID ! ID optional_minus INTEGER ! ID STRING 31. 32. 33. synonyms = 'SYNONYMS' synonym_list ; 34. synonym_list = synonym 35. ! synonym_list synonym 36. synonym = id_list '=' ID '.' ; 37. messages = 'MESSAGE' message_list ; 38. message_list = message ! message_list message 39. ; 40. message = ID ':' STRING ; 41. syntax = 'SYNTAX' syntax_list ; 44. syntax_item = ID '=' syntax_elements optional_class_restrictions 45. syntax_elements = syntax_element
46. ! syntax_elements syntax_element ; 49. optional_multiple_indicator = ! '*' 50. 51. optional_class_restrictions = '.' . ! 'WHERE' class_restrictions 52. 53. class_restrictions = class_restriction ! class_restrictions 'AND' class_restriction 54. ; 55. class_restriction = ID 'ISA' classes 'ELSE' statements 56. classes = class_identifier
57. ! classes 'OR' class_identifier 57. ; 58. class_identifier = 'OBJECT'
59. ! 'ACTOR' 60. ! 'CONTAINER' 61. 'INTEGER' . ! 'STRING' 62. ! 'CONTAINER' 'OBJECT' ! 'CONTAINER' 'ACTOR' 63. 64. 65. optional_verbs = 66. ! optional_verbs verb ; 67. verb = verb_header verb_body verb_tail 68. verb_header = 'VERB' id_list

;

```
69. verb_body = simple_verb_body
70. ! verb_alternatives
;
73. verb_alternative = 'WHEN' ID simple_verb_body
74. simple_verb_body = optional_checks optional_does
75. verb_tail = 'END' 'VERB' optional_id '.'
76. optional_checks =
77. ! 'CHECK' statements
78. ! 'CHECK' check_list
.
80.
81. check = expression 'ELSE' statements
83.
84. location = location_header location_body location_tail
85. location_header = 'LOCATION' ID optional_name
87. location_tail = 'END' 'LOCATION' optional_id '.'
              •
88. optional_exits =
              ! optional_exits exit
89.
90. exit = 'EXIT' id_list 'TO' ID optional_exit_body '.'
optional_id
93. object = object_header object_body object_tail
94. object_header = 'OBJECT' ID optional_name optional_where
95. object_tail = 'END' 'OBJECT' optional_id '.'
             ;
97. optional_attributes =

<sup>98</sup> ! optional_attributes is attributes
99. is = 'IS'
100. ! 'ARE'
```

```
101. ! 'HAS'
       ;
102. optional_descriptions =
                         ! optional_descriptions
103.
                          article_or_mentioned_or_description
                         .
104. article_or_mentioned_or_description = article
                                      ! description
105.
                                      ! mentioned
106.
107. optional_description =
             ! description
108.
                        ;
109. description = 'DESCRIPTION'
110. ! 'DESCRIPTION' statements
;
111. article = 'ARTICLE'
112. ! 'ARTICLE' statements
;
113. mentioned = 'MENTIONED' statements
             ;
114. optional_name =
115. ! 'NAME' ids
.
118. container = container_header container_body container_tail
119. container_header = 'CONTAINER' ID
120. container_body = optional_limits optional_header optional_empty
121. container_tail = 'END' 'CONTAINER' optional_id '.'
                  ;
124. limits = limit
125. ! limits limit
;
126. limit = limit_attribute 'THEN' statements
          ;
127. limit_attribute = attribute
          ! 'COUNT' INTEGER
128.
                   ;
129. optional_header =
130. ! 'HEADER' statements
                   ;
131. optional_empty =
      ! 'ELSE' statements
132.
133. event = event_header statements event_tail
          ;
134. event_header = 'EVENT' ID
135. event_tail = 'END' 'EVENT' optional_id '.'
```

```
136. actor = actor_header actor_body actor_tail
137. actor_header = 'ACTOR' ID optional_name optional_where
138. actor_body = optional_properties optional_attributes
                  optional_description optional_verbs optional_actor_script
139. actor_tail = 'END' 'ACTOR' optional_id '.'
140. optional_actor_script =
! optional_actor_script script
142. script = 'SCRIPT' INTEGER '.' optional_description step_list
143. step_list = step
         ! step_list step
144.
                 ;
145. step = 'STEP' statements
146. ! 'STEP' 'AFTER' INTEGER statements
147. ! 'STEP' 'WAIT' 'UNTIL' expression statements
148. rule = 'WHEN' expression '=>' statements
149. start = 'START' where '.' optional_statements
150. optional_statements =
                      ! statements
151.
                              ;
152. statements = statement
153.  ! statements statement
;
                   ;
154. statement = output_statement
        ! special_statement
! manipulation_statement
155.
156.
                 ! event_statement
! assignment_statement
157.
158.
                 ! actor_statement
159.
160.
                 ! if_statement
161. output_statement = STRING
                          ! 'DESCRIBE' what '.'
! 'SAY' expression '.'
! 'LIST' what '.'
162.
163.
164.
                          ;
! 'SAVE' '.'
! 'RESTORE' '.'
167.
168.
                            ! 'SCORE' optional_integer '.'
! 'VISITS' INTEGER '.'
! 'SYSTEM' STRING '.'
169.
170.
171.
172. manipulation_statement = 'EMPTY' what optional_where '.'
173. ! 'LOCATE' what where '.'
174. event_statement = 'CANCEL' ID '.'
                         ! 'SCHEDULE' ID optional_where 'AFTER' INTEGER '.'
175.
                         ;
176. assignment_statement = 'MAKE' what something '.'
177. ! 'INCREASE' attribute_reference optional_by_clause
                                 '.'
```

```
178.
                              ! 'DECREASE' attribute_reference optional_by_clause
                                     'SET' attribute_reference 'TO' expression '.'
179.
                                   ı.
180. optional_by_clause =
                                ! 'BY' expression
181.
                                ;
182. if_statement = 'IF' expression 'THEN' statements optional_elsif_list
                         optional_else_part 'END' 'IF'
                                                                    ۱. ۱
                       ;
183. optional_elsif_list =
            _ _ ! elsif_list
184
;
187. optional_else_part =
188. ! 'ELSE' statements
                                ;
189. actor_statement = 'USE' 'SCRIPT' INTEGER optional_for_actor '.'
                            ;
190. optional_for_actor =
191.
                               ! 'FOR' ID
192. expression = term
193. ! expression 'OR' term
194. term = factor
195. ! term 'AND' factor ;
196. factor = primary
197. ! primary optional_not where
198. ! primary binop primary
199. ! primary optional_not relop primary
200. ! primary is something
201. primary = optional_minus INTEGER
202. ! STRING
203. ! what
204. ! 'SCORE'
205. ! aggregate where
206. ! '(' expression ')'
207. ! attribute_reference
208. ! 'RANDOM' INTEGER 'TO' INTEGER
;
                 ;
209. aggregate = 'COUNT'
210. ! 'SUM' 'OF' ID
211. ! 'MAX' 'OF' ID
                   ;
212. something = optional_not ID
213. what = 'OBJECT'
214. ! 'LOCATION'
215. ! 'ACTOR'
216. ! ID
             ;
217. optional_where =
218. ! where ;
219. where = 'HERE'

      220.
      ! 'NEARBY'

      221.
      ! 'AT' what

      222.
      ! 'IN' what
```

```
223. binop = '+'
224. ! '-'
225. ! '*'
226. ! '/'
;

227. relop = '<>'
228. ! '='
229. ! '=='
230. ! '<='
231. ! '<='
232. ! '>'
233. ! '<'
;

234. optional_qual =
235. ! 'BEFORE'
236. ! 'AFTER'
237. ! 'ONLY'
;

238. optional_not =
239. ! 'NOT'
;

240. optional_id =
241. ! ID
243. ! ids ID
243. ! ids ID
243. ! ids ID
244. id_list = ID
245. ! id_list ',' ID
;

246. optional_integer =
247. ! INTEGER
;

248. optional_minus =
249. ! '-'
;

250. attribute_reference = ID 'OF' what
;
</pre>
```

;

C COMPILER ERROR MESSAGES

This appendix describes the error messages generated by the Alan compiler. For each message, a short description of the error, possible causes etc. are given. Each message reported also indicates the severity of that error. The message is supplemented with an indication of its severity. An informational message (indicated by an I) simply gives some information to the user, a warning message (W) indicates an error but the compilation still generates a valid output (although not always what the user intended). Error messages (E) indicate errors which have made it impossible to generate any output, but the compiler continues to process all input. Fatal (F) and system (S) messages always terminate the compilation process immediately.

The message descriptions below may also contain the special insertion markers \n' (where n is a digit), which indicate that text will be inserted during compile time at that point in the message, e.g. the offending identifier or a file name.

100 Parsing resumed here.

A severe syntax error was discovered. Some input was skipped. This error message marks the place where the parsing was restarted.

101 %1 inserted.

A syntax error was discovered and one or more symbols inserted in the input in an attempt to recover.

102 %1 deleted.

A syntax error was discovered and one or more symbols was skipped from the input in an attempt to recover.

103 %1 replaced by %2.

A syntax error was discovered and one or more symbols was replaced by one or more other symbols in an attempt to recover.

104 Severe syntax error, construct ignored.

An intricate syntax error was discovered. A complete construct was skipped in an attempt to recover.

- 105 Syntax error, couldn't recover.
- 106 Parse stack overflow.
- 107 Parse table error.
- 108 Parsing terminated. Alan compiler implementation errors. Should not occur!

150 Unterminated STRING.

An opening double quote was not terminated by a closing quote before end of file. Error message points to the opening quote. Remember STRINGs may cover several lines.

151 File name missing for \$INCLUDE directive.An include directive was given but the file indicated was not found.

The complete file name must be given.

198 Could not open output file '%1' for writing.

The indicated output file could not be opened, probably because the directory did not exist, the file or directory was write protected.

199 Adventure source file (%1) not found.

The source file given on the command line did not exist. The Alan compiler adds the **.alan** extension to the file name given, if it did not include a period.

201 Mismatched block identifier, '%1' assumed.

The identifier following a terminating END did not match the one given at the beginning of the construct. This indicates an illegal nesting or a missed END IF. The identifier indicates to which block the END is assumed to belong.

- 202 Multiple usage of direction '%1' in this EXIT.
- 203 Multiple definition of EXIT '%1' in this location.

The directional word indicated was used more than once, either in the same, or different exit declaration from the location. This is contradictory and not legal.

- 204 Multiple definition of %1 DEFAULTS. Ignored.Only one declaration of default attributes per type is allowed. The second declaration is ignored.
- 205 Multiple usage of '%1' in this VERB definition.

When specifying actions for multiple verbs in the same declaration, the indicated word occurred twice.

206 Multiple definition of SYNTAX for %1.

More than one syntax definition for the same verb was found. This is an error, remove the offending one. 207 VERB '%1' is not defined. A SYNTAX construct defined the syntax for a verb that was never defined.

208 '%1' is not a VERB.

The identifier on the left hand side of a SYNTAX definition was defined as something that was not a VERB.

- 209 First element in a SYNTAX must be a player word. The definition of a SYNTAX may not start with a parameter. The first word must be a player word so as to distinguish it from other forms of input.
- 210 Action qualification not allowed here. The BEFORE, AFTER and ONLY qualifiers may not be used for a DOES-clause in this context.
- 211 Adventure must start at a Location.

You specified a where expression in the START section that did not specify an explicit location. The start section specifies where the hero starts and must be a LOCATION.

212 Syntax parameter '%1' overrides symbol.

The SYNTAX definition valid in this context defined a symbol which is the same as an entity (object, location or actor). The syntax parameter will take precedence.

213 Verb alternatives not allowed here.

You may only specify different verb body alternatives within objects. The global verb body and the verb body in the location may not have alternatives.

214 Parameter not defined in syntax for '%1'.

The identifier given as the selector in a verb body alternative was not defined in the syntax for that verb.

215 Syntax not compatible with syntax for '%1'.

To be able to use the same body for different verbs by supplying them in a comma separated list in the verb header they must all be compatible. This means that they have the same number of parameters and the parameters have the same names. Otherwise conflicts will arise when figuring out which parameter to use.

216	Parameter	'%1'	multiply	defined	in	this	SYNTAX.	

The parameter was defined more than once in the same SYNTAX definition.

217 Only one multiple parameter allowed for each syntax. This one ignored.

To be able to use multiple parameters in a player command only one parameter may be marked as referring to multiple objects or actors using ALL or conjugations. This is a warning, the syntax will be as if the first multiple marker was the only one.

218 Multiple definition of attribute '%1'.

The indicated attribute name was defined more than once in the same context (default attribute list or within the same entity). Remove one definition.

220 Multiple definition of '%1'.

The indicated word has multiple, and possibly different, definitions.

221 Multiple class restriction for parameter '%1'.

The same parameter occurred more than once in the list of class restriction in the same SYNTAX definition.

- 222 Identifier '%1' in class definition is not a parameter. Only the parameters in the syntax may be referenced in the class restricting clause of a SYNTAX definition.
- 230 No syntax defined for this verb, assumed '%1 (object)'. This message is a warning to indicate that the default syntax handling has been used.
- 310 Identifier '%1' not defined.

The indicated word was never defined. It must be declared as either a location, an object, a container, an actor or an event.

311 Must refer to %1.

The construct indicated does not refer to the correct kind of item, the message indicates which kind of item was expected.

312 Parameter not uniquely defined as %1, which is required. In certain contexts it is necessary to refer to a particular type of entity, e.g. the IN expression must refer to a container or an object with the container property. If the reference (the what part) is a parameter identifier, this parameter must be restricted to be of the required type by use of parameter restrictions (such as 'WHERE c ISA CONTAIN-ER').

- 315 Attribute not defined for '%1'.The indicated attribute is not defined for the particular object, location or actor. It must either be a default attribute or be locally declared.
- 318 Entity '%1' is not a Container.

The referenced entity (object or actor) was not declared to have the container property, although the context required a container.

320 Word '%1' belongs to multiple word classes.

A word was declared as to belong to different word classes such as noun, synonym, adjective etc.

321 Synonym target word '%1' not defined.

To define a synonym its target word (the word on the left side of the equal sign) must be defined as a proper word elsewhere in the source.

322 Word '%1' already defined as a synonym.

A word may not be declared as a synonym for different target words.

330 Wrong types of expression. Must be of %1 type.

In an expression, a value or an expression was used that had a type that was not allowed. The message indicates the correct type.

331 Incompatible types in %1.

The two values in an expression with a binary operator did not have compatible types, or the value used in a SET statement was not type compatible with the referenced attribute.

- 332 Type of local attribute must match default attribute. An attribute declared locally (within an object, actor or location) that has the same name as a default attribute, has to have the same type (boolean, integer or string).
- 333 The word '%1' is defined as a synonym as well as of another word class.

Synonyms must be words *not* defined elsewhere.

400 Script not defined for Actor '%1'.

No script with the indicated number was defined for the actor.

401 Actor reference required outside Actor specification.

Inside an actor specification it is permissible to leave out the actor reference in a USE statement in which case the surrounding actor is assumed. Outside actor specifications, the actor reference must always be supplied.

402 An Actor can't be inside a Container.

The LOCATE statement tried to locate an actor inside a container. This is not allowed.

- 403 Script number multiply defined for Actor '%1'. The indicated number was used for more than one script for the same actor.
- 404 Attribute to %1 must be a default attribute.

To reference attributes for OBJECT, LOCATION and ACTOR the attribute used must be a default attribute, as all objects, locations or actors must have it.

405 The class of a parameter used in %1 must be uniquely defined.

In some statements the class of the identifier must be determined during compile time. This is, for example, the case in MAKE and SET statements.

406 A parameter defined as Container have no default attributes.

A parameter that was restricted to containers do not have any default attributes. Actors, objects and locations have separate sets of default attributes. In order to refer to an attribute on a parameter it must be restricted to one of these classes. If the parameter also requires the container property, use CONTAINER ACTOR or CONTAINER OBJECT.

- 407 Attribute in LIMITS must be a default attribute. All objects must have the attribute that a limit is to test.
- 408 Attributes in %1 must be of boolean type. The attribute referenced in the indicated context must be a boolean attribute.
- 409 No parameter defined in this context.

No parameter is defined in the context where a reference to OBJECT was made. Parameters are only defined within checks and bodies of verbs, so the use of OBJECT (an obsolete construct, use the parameter identifier instead) is also restricted to those contexts. See *Run-time Contexts* on page 60.

- 410 A parameter may not be used in %1.In certain statements a parameter may not be used at all.
- 411 %1 ignored for Actor 'hero'.

It is allowed to redefine the predefined actor HERO (the player) which makes it possible to define local attributes and descriptions for the hero. However any definition of scripts or initial location is ignored (the script is supplied by the player in his input and the initial location is defined in the START section).

412 'ACTOR' is not allowed inside events.

In events no actor is active. This means that no reference to the active actor using ACTOR can be made.

- 413 Expression in %1 must be of integer type. The context required a numeric expression.
- 414 Invalid initial location for %1.The initial location specified was not valid.
- 415 Invalid Where specification in %1 statement. The statement indicated does not allow the where specification used.
- 416 Interval of size 1 in RANDOM expression.

This message informs that the interval in a RANDOM statement was just one single value, resulting in always returning the same value. Not very random.

417 Comparing two constant entities will always yield the same result.

The expression compared two identifiers none of which was a parameter. This will always give the same result. This is probably an error, but the message is still a warning as it gives a perfectly running adventure (but, perhaps not what you intended?).

- 418 Aggregate is only allowed on integer type attributes. The aggregates MAX and SUM can only perform their calculation on integers.
- 419 Expression in %1 must be of integer or string type. In the indicated context only integer and string type expressions may be used.

501 LOCATION '%1' has no EXITS.

In case the hero is located at the indicated location he may not be able to escape from that location. This may be intentional (as for a limbo location or a location with magic words to use as an escape) but the warning is presented as a reminder.

600 Multiple use of option '%1', ignored.

The indicated option was used more than once, this occurrence is ignored and the previous setting used.

601 Unknown option, '%1'.

A word was given in the option section that was not the name of an option.

602 Illegal value for option '%1'.

The indicated option does not allow the value used.

997 SYSTEM ERROR: %1

A severe implementation dependent error has occurred (a bug!). Please report.

- 998 Feature not implemented in %1. The combination of some syntactically correct but semantically tricky constructs are not yet implemented. Please report.
- 999 No Adventure generated.

When an error is detected this informational message is given to indicate that no executable adventure was output.

D HOW TO USE THE SYSTEM

D.1 Compiling

This version of the Alan Adventure Development System is a traditional batch compiler. This means that the actual development system is a compiler which read text files created using any normal text editor. To compile an adventure use the command:

alan <adventure>

where <adventure> is the name of the main file containing your adventure source text. The compiler will add an extension, **.alan** (or **.ala** on PCs), if none is supplied. The option **-help** will give a brief help on other options to the compiler.

The output from the compiler, **alan**, is two files, an adventure code file *adven-ture*.acd and an adventure data file, *adventure*.dat.

D.2 Compiler Switches

The compiler supports the following switches:

- -verbose prints compiler version and other verbose messages
- -warnings show warning messages from the compilation process
- **-infos** show informational messages from the compilation process
- **-full** give a complete listing of the source on the screen
- **-height <n>** use page height <n> (lines) when producing list files
- **-width** <**n**> use page width <**n**> (columns) when producing list files
- **-debug** include debugging information in the produced adventure files (same as the debug option, see *Options* on page 21)
- **-pack** encode and compress the text data (same as the pack option, see *Options* on page 21)
- **-summary** produce a summary about number of actors, size of adventure files, timing information etc.
- **-dump** print the internal form (developers use only)

Giving an extra hyphen before the option reverses its meaning, e.g. --warnings means don't show warnings. Switches may be abbreviated.

D.3 Running the Adventure

To play the generated adventure the Alan interpreter, **arun**, is executed with the adventure name as a parameter.

```
arun <adventure>
```

No extension on the adventure name is allowed.

If the interpreter program is copied to a different name it will look for code and data files with the same name. Any parameters or switches will be ignored. For example, by copying the **arun** program to **adventure** the interpreter will, when started under the new name, directly look for the files **adventure.acd** and **adventure.dat**. The three files **adventure**, **adventure.acd** and **adventure.dat** thus makes a complete game package which is easy to start using the single command:

> adventure

D.4 Interpreter Switches

The interpreter supports the following switches:

- -d print interpreter version and enter debug mode
- -i ignore CRC and version errors in the adventure files
- **-t** trace sections executed
- -s show single instruction trace

In later versions an interactive development environment is envisioned but this is still far away. So you have to be content with the debugging support described in *Debugging* on page 71 for now.

E SYSTEM DETAILS

A complete Alan system should contain the following files

alan - executable image for Alan compiler arun - executable image for Alan interpreter jungle.alan dungeon.alan COPYING - description of copying conditions alan.ps - postscript version of this document alan.txt - plain text version of this document

Depending on the environment the two executable files may have different names, such as **alan.exe**.

E.1 Amiga

The Alan system is delivered as an **lha** archive. To install the Alan system, create a directory and copy the archive file (e.g. **alan.lha**) there. Use the commands

lha x alan.lha lha x arun.lha

(or equivalent depending on archiver) to extract the files. Include this directory in your path. The easiest way to do this is to include a command like

path SYS:Alan

in your **startup-sequence** (under Workbench 2.x or above, **s:User-startup**) and then reboot. Another possibility is to type the above command in a Shell (which will only make these definitions available until next reboot).

After this the Alan system is ready for use. The example adventure (see appendix E, *SYSTEM DETAILS*, on page 103) is included in the delivery and you may compile and run it with the commands

```
alan example
arun example
```

The Alan compiler requires more than the standard stack size (4096), a size of 20000 has been used without trouble.

The Alan interpreter **arun** supports Workbench-startup through double-clicking on the Arun-icon. The tooltype WINDOW is supported to make it possible to selecting the window in which the adventure should be run. If a console handler device such as NEWCON: in 1.3 or the normal CON: in 2.x and above history and command line editing is available.

E.2 Unix

On UNIX systems the Alan system is delivered in a compressed **tar** archive. To extract the files give the commands:

```
uncompress alan.tar.Z
tar xvf alan.tar
```

You may copy the executable files, **alan** and **arun**, to a directory which is in your path or include the relevant directory in the PATH environment variable. The installation may be tested by giving the commands

alan example arun example

which should compile and run the example adventure (see appendix I, EX-AMPLE ADVENTURE, on page 80).

On UNIX systems command history, recall and editing is available.

E.3 MS-DOS and OS/2

For MS-DOS and OS/2, the Alan system is delivered in a **pkzip** compatible **zip** archive. To extract the files issue the commands

unzip alan.zip unzip arun.zip

The current directory will contain the executables as well as the samples and documentation. Make sure this directory is included in your PATH variable.

Testing may be performed as above.

E.4 Portability of Games

The adventure files produced by the compiler is compatible across all supported platforms. This means that by copying the binary **.acd** and **.dat** files to another machine they should be possible to interpret by an interpreter on that new machine without any changes. Note however that the files must be transferred in *binary* mode (where applicable). Also the character sets are automatically converted to the native set allowing multi-national characters to be presented correctly even on machines that does not support the ISO 8859-1 standard. This is of course restricted to characters having a representation in the current native character set.

It is a strong goal to achieve complete portability of the games to be able to provide games for all supported platforms without re-compilation. Game authors should take this into serious consideration when designing games and not use any system specific characters, character combinations or special commands that may be available on some systems.

F DIFFERENCES BETWEEN VERSIONS

Version 2.6

The 2.6 interpreter will run 2.5 games, but the 2.6 compiler can not generate 2.5 games. So upgrading to 2.6 will create games only playable with 2.6 interpreters, but you can keep old games and still play them.

User definition of run-time messages is now possible.

Removed the indefinite article from the default messages. Instead introduced the ARTICLE slot in objects which will be used (if present) before producing the MENTIONED message (which may be constructed automatically). If no article is declared a default is supplied ("a" if using english). This means that some tricks that have been used to somewhat remedy the article problem ('a' was always used!), don't work any more. Remove all 'a', 'an' etc. from the texts and names in the Alan source (usually in the MENTIONED slot and possibly in the HEADER for containers), and introduce the ARTICLE "an" declaration on objects that require it (those whose name start with a vowel sound). For objects that doesn't need an article define an empty ARTICLE clause.

It also means that their is now a new reserved word ARTICLE.

It is also now possible to define the ARTICLE, MENTIONED and DESCRIP-TION on objects in any order.

Version 2.5

String quotes (") within strings are now allowed, if doubled ("Charlie said ""Hello!""). The same goes for single quotes (') within quoted identifiers.

(Luis Torres <let@reef.cis.ufl.edu>)

Multiple default attribute sections simplifies using general include files as the default attributes can be distributed across the complete adventure source.

The new VISITS statement replaces the previous option with the same name, allowing setting of the visits variable during run-time.

The compiler now generates completely compatible adventure files, including multi-national character sets which are converted automatically to be presented correctly on any supported platform.

If the interpreter is renamed it will automatically load adventure files (.acd and .dat) with the same name.

(Jeff Harrison <harrison@mprgate.mpr.ca>)

The QUIT statement prints a restart question which may be answered affirmative, in which the game is reloaded and restarted, or negative in which case the adventure is terminated.

(Byron Montgomerie <byron@saturn.cs.mun.ca>)

SAVE and RESTORE now prompts for a file to use so multiple save files can be used by the player.

(Luis Torres <let@reef.cis.ufl.edu>)

Multiplication and division can now be performed using the '*' and '/' operators respectively.

(Robert Yoke-Loong Foo <af685@freenet.carleton.ca>)

Version 2.4

Actors may now be containers (allows for making them carry things). The class indications in the syntax declarations have been enhanced to account for this also. You can now restrict parameters to all entities having the container property, only actors having it, or only objects having it (see *Syntax Definitions* on page 25, and *Containers* on page 36 for details).

String comparison normally ignores the case of characters (the new operator '==' does exact matching) (see *Binary and Relational operators* on page 51).

The statements to increase or decrease values are now called INCREASE and DECREASE (instead of INCREMENT and DECREMENT).

An optional description has been introduced on actor scripts, giving a possibility to create descriptions that are directly coupled to the activities of the actor (refer to *Actors* on page 38).

The QUIT statement now does not print any scores. This has to be made explicitly. Also the identifier SCORE is now allowed in expressions, representing the current value of scores collected so far.

Containers are now listed in a more natural way, the old format of one item per line has been replaced by concatenating them into a natural sentence, like:

You are carrying a box, a ball and a lightbulb.

This might require a change to the HEADER declaration of containers.

Version 2.3

String and integer literals are introduced in the player input and in the syntax declarations. Attributes may now also be strings. No incompatibilities should occur.

Version 2.0

In version 2, the concept of syntax is introduced. A programmer may allow different and more complex input from the player, not just the simple verb/ object type used in version 1. However, the default mechanism is still this simpler form of input so very little needs to be changed when converting to version 2. This also follows the spirit of Alan, it means that syntax is not strictly necessary unless you want to do something extra. For player input following the simple verb/object syntax there is nothing you have to do.

Another difference is the improvement in the definition of synonyms. First, the order of definition is different, you should now supply all the synonyms first and *then* the word they are synonyms for. This will probably require some rewriting of your Alan programs, but it is the more logical way to specify synonyms. Also, synonyms are now allowed anywhere in the program, so it is now possible to group global verb definitions, syntax definitions and synonyms for the same verb together (and perhaps place them in a separate include file).

G FUTURE DEVELOPMENTS

As Alan is an application oriented language, i.e. it is designed to fit a particular application domain perfectly (in this case adventure authoring), it is dependent on adventure authors requirements and ideas for its further evolution. So please let us know!

email:	thoni@softlab.se gorfo@ida.liu.se		
postage:	Thomas Nilsson Stenbrötsgatan 57 S-582 47 LINKÖPIN SWEDEN	phone: G	Int. +46 13 12 11 67 Nat. 013 - 12 11 67
	Göran Forslund Vallmogatan 22 S-582 46 LINKÖPIN SWEDEN	phone: G	Int. +46 13 13 39 91 Nat. 013 - 13 39 91

But here are some ideas:

- Definition of common attributes, verb definitions etc. through introduction of a class structure.
- Definition of interaction with actors, perhaps through some kind of pattern matching sub-language using string literals.
- Background pictures
- Sound
- ...

H REFERENCES

[Ada80] Scott Adams : *Pirate's Adventure*; BYTE December 1980, pp 192-212 An article describing the history behind the Scott Adam's adventures, particularly the *Pirate's Adventure*. Also includes BASIC source for the adventure, consisting mostly of DATA-statements.

[Bla80] Marc S. Blank, S. W. Galley : How to Fit a Large Program Into a Small Machine; Creative Computing July 1980, pp 80-87 A good article on the internals of the Z-interpreter, the pseudo-machine created by Infocom for creating and running adventures. As always from the hands of the Infocom men, also very good reading.

[Bet87] David Betz : An Adventure Authoring System; BYTE May 1987, pp 125-135 A description of a system similar to Alan, AdvSys, consisting of a special purpose language, a compiler and an interpreter for it. At last the term *authoring* is used instead of *programming*. The system is available through various PD-sources such as Fred Fish, BIX etc.

 [Bra84] A. J. Bradbury : Adventure Games for the Commodore 64; Granada Publishing 1984, ISBN 0-246-12412-1
 A good book, especially on the topic of adventure writing methodology. Carries the concept of storyboarding a bit further than [Gra83]. Also contains interspersed utilities and modules (in C64 BASIC) and a small adventure, "The Case of the Lost Adventure".

- [Bri84] Tony Bridge, Richard Williams : *Sinclair QL Adventures*; Sunshine Books 1984, ISBN 0-946408-66-1 Contains a few good chapters on adventures and reviews of some games of the classical text-type, but then goes on to present the listing of a fairly uninteresting "adventure generator" for a menu-driven *Dungeon And Dragons* inspired (much fighting, strength scoring and banes and such) kind of adventures games.
- [Buc87] Mary Ann Buckles : *Interactive Fiction as Literature*; BYTE May 1987, pp 135-142 A very interesting article discussing the literary heritage of adventure games and their future in that perspective.
- [Fic86] Erik Fichtelius : Nu kommer det svenska äventyrsspe-

let!; Upp&Ner, nr 2 1986 A swedish article describing the famous swedish "Stuga" game, created around 1980, which at that time was available for the PC.

[Gra83] Mike Grace : Commodore 64 Adventures; Sunshine Books 1983, ISBN 0-946408-11-4 A fairly good book on playing and writing adventure games, written by an beginner programmer. Strictly BASIC programming but contains many good ideas to borrow. Includes some short sections on methods and mentions the concept of storyboarding. Contains a type-in adventure ("Nightmare Planet") for the C64.

[Geu85] A.F. de Geus, J.H. Jongejan, A.M. Koelmans : *Adventure Description Language*; Sigma Press 1985, ISBN 1-85058-011-1 Describes an assembler-like Adventure Language for the BBC Micro, and uses its design as a vehicle for briefly describing a few basic computer science techniques (e.g. grammars, hashing, huffman coding and graph theory). Source (in ADL!) for "Red Button" and "Long Forgotten Arabia" adventures plus complete source for the "scanner", "interpreter" and "editor" for ADL. Note: this is *not* the better known ADL by Ross Cunniff.

- [Goe93] Phil Goetz : *Interactive Fiction*; Dept. of Computer Science, SUNY, Buffalo NY 14260, USA Interesting paper setting out to define the term interactive fiction. Also discusses history and future of IF, and various media it may use.
- [Gra87] David Graves : Second Generation Adventure Games; Journal of Computer Game Design, Volume 1, number 2 (August 1987), pp 4-7 An article describing many of the more fundamental concepts (conceptual and implementational) of interactive fiction of today, such as object orientation, natural language, text generation and goal orientation.
- [Gra88] David Graves : Bringing Characters to Life; Journal of Computer Game Design, Volume 2, number 2 (December 1988), pp 10-11 Describes the role and implementation of artificial personalities in interactive fiction. This feature is seldom implemented in main stream interactive fiction but would probably give greater depth to the nonplayer characters in the story.
- [Gra91] David Graves : *Plot Automation*; Journal of Computer Game Design, Volume 5, number 1 (October 1991), pp 10-12

The interesting idea of automatically creating a plot from the personalities and goals of the actors in the story is presented and discussed.

- [Het84] Tony Hetherington : Adventure Games; Personal Computer World, January 1984 (October 1991), pp 17-26
 Introductory discussion on what makes a good adventure, text vs. graphic, then some reviews on current games, e.g. The Hobbit and Snowball.
- [Has80] Greg Hassett : *How to write An Adventure*; Creative Computing July 1980, pp 88-90 A short superficial article containing nothing that can't be found elsewhere.
- [Leb79] P. David Lebling, Mark S. Blank, Timothy A. Andersson : *ZORK - A Computerized Fantasy Simulation Game*; IEEE Computer, April 1979 An interesting article describing the inner workings and motivations behind ZORK by the men who (almost) started it all.
- [Leb80] P. David Lebling : ZORK and the Future of Computerized Fantasy Simulations; BYTE December 1980, pp 172-182
 Lebling again describes the Zork world and machine. This article adds discussions on various implications of continuing to development, such as intelligent actors and communication with them, how far to take the parsing of natural language and how careful you must be before adding another feature in the games universe.
- [Lid80] Bob Liddil : On the Road to Adventure; BYTE December 1980, pp 158-170 Some tips for playing and reviews of number of not so famous adventures (by Adams, Hassett, Programmer's Guild and Mad Hatter).
- [Mit86] David Mitchell : An adventure in programming techniques; Addison-Wesley 1986, ISBN 0-201-15030-1 An excellent book covering almost every aspect of adventure playing and writing. As the title suggests adventure writing is taken as the goal for presenting various programming techniques, but still with the problems of writing and designing adventures as the primary issue. A bible for adventurers.
- [McG84] Gary McGath : *COMPUTE!'s Guide To Adventure Games*; COMPUTE! Books 1984, ISBN 0-942386-67-1 An excellent book, its primary merit is the reviews of

most of the Infocom adventures, all Scott Adam's and a bunch of various other adventure games available and popular in 1984. Also contains a field guide for adventurers and a short discussion on how to program your own games. Includes source (in various dialects of BASIC!) for "Tower Of Mystery". The concluding chapter on the future of adventure games is most intriguing and may serve as a source for inspiration when trying to push its limits.

- [Owe83] Peter Owens : Adventures in Learning; Popular Computing, December 1983, pp. 147-150 An article discussing how computer games, adventures in particular, can be used in education and their potential effect of learning people to think.
- [Sca81] Peter D. Scargill : Adven-80, An Advanced Adventure Development System; Dr. Dobb's Journal, Number 61 (November 1981) An interesting predecessor, assembler like in structure with a lot of "magic numbers", but was probably a good system at the time.

EXAMPLE ADVENTURE T

This section contains a small example of how an adventure can be written in Alan. The emphasis have not been on the ultimate features of the language. Instead it is intended to show how much functionality can be achieved by just a few hundred lines of code.

-- This is an example of an adventure written in ALAN using almost

-- nothing of the more advanced features.

-- The story is not much: You have lost your memory and stumble around

-- on a narrow path in the middle of the jungle. To the north the path -- takes you to a river and to the south to a clearing where a tiger -- blocks your way. The only way to get past the tiger is to eat a -- certain kind of fungus, which works as tiger repellant (a clue about

-- this can be found in your notebook). The fungus can only be found

-- by climbing the vine hanging down over the path. When you have

-- succeded in getting past the tiger the game gets to a happy ending.

LOCATION Path

DESCRIPTION

"You are standing on a barely visible path in the middle of nowhere The path looks like it's been walked by bare feet (or rather paws) for many a year. From the small amount of light reaching the ground here I should say the path runs in almost straight north/south direction. On both sides of the path is the deepest, darkest jungle you've ever seen. I really wouldn't recommend going that way. The path itself isn't much of a place to hold on to either. You get the impression that the vegetation is trying hard to recapture even this tiny part of land. The trees on both sides seems to come closer and there are vines hanging down almost touching your head.'

EXIT north TO bank. EXIT south TO clearing. EXIT east, west TO jungle. END LOCATION.

LOCATION Bank

DESCRIPTION

"The path ends here on the south side of a wide river. On the ground you can see lots of paw prints (some pretty big ones, too). The obvious guess is naturally that this is a common place for the wild animals to stop by for a drink or two (and maybe a bite too). The river itself doesn't seem to be too dangerous - it's neither too wide nor too rapid but those logs with a pair of eyes give you second thoughts.

EXIT north, swim TO river. EXIT south TO path. EXIT east, west TO jungle. END LOCATION.

LOCATION Trees

DESCRIPTION "You have now ended up high above the ground in the middle of the trees and vines. The vegetation is so thick up here that it seems almost like a green floor.'

EXIT down TO path. END LOCATION.

LOCATION River

DESCRIPTION

"Defying the obvious horrors of the river you try for the northern river bank. One crocodile immediately chops your left foot of, but you makes it almost to the middle of the river before another merciful crocodile finish you off."

END LOCATION. _____ LOCATION Clearing DESCRIPTION "Here the jungle opens up a bit and the path takes you straight into a clearing. The path seems to continue on the south side of the clearing some fifty paces away." EXIT north TO path. EXIT east, west TO jungle. EXIT south TO camp CHECK hero IS repelling ELSE "The tiger opens its big mouth and lets out a terrifying growl. Apparently it won't let you pass." DOES "When you approaches the tiger it looks confused. Then it really takes in your smell. It suddenly bolts, turns and takes off into the jungle." LOCATE tiger AT nowhere. END EXIT. END LOCATION. LOCATION Jungle DESCRIPTION "Now you've really done it. Didn't I tell you NOT to enter the jungle." EXIT north, south, east, west TO jungle DOES "Stumbling around in the jungle trying to make your way through the damp vegetation that almost seems to reach out for you, you suddenly stumble onto a snake, which disapprove very clearly of you stepping on it. One bite in the leg and you have had it." OUIT. END EXIT. END LOCATION. _____ LOCATION Camp DESCRIPTION "Here is the scattered parts of what ones was the camp of your expedition. The sight of it makes your memory come back. When you were attacked last night of a herd of wild elephants everyone fled in panic. You yourself ran straight into a tree and must have lost both conciousness and memory. 'Well, hope the computer still works.' you think. 'I think I stick to computer adventures, at least for the immediate future.'" OUIT. END LOCATION. LOCATION nowhere -- The location for disappearing objects. END LOCATION. _____ OBJECT Tiger AT Clearing DESCRIPTION "An enormous tiger is standing here blocking your way." END OBJECT. OBJECT Notebook IN inventory DESCRIPTION "The book is called 'The Jungle Book: Tricks and Tips'. It also has your name on it.' VERB Take DOES LOCATE OBJECT IN inventory. "Taken!" END VERB.

OUIT.

VERB Drop DOES

```
LOCATE OBJECT HERE.
          "Dropped!"
    END VERB.
    VERB Read DOES
         "You open the book and glance over the notes. It is really
         a very strange mixture. Something about a tree you shouldn't
hide under when it rains, 'cause some kind of bugs will start
falling of its leaves, something else about a certain kind of
fungus, which grows up among the vines and when eaten is a
strong tiger repellant and something about how to make a fire
         from wet moss. Here are page after page of useful hints of
         how to survive in the jungle, all in your own hand writing."
    END VERB.
END OBJECT.
OBJECT Vine AT Path
    DESCRIPTION
         "A particulary long and thick vine is hanging down just beside
         you.'
    VERB climb DOES
          "The vine is quite slippery, but you still manage to climb
         well into the trees."
         LOCATE HERO AT Trees.
    END VERB.
END OBJECT.
OBJECT Fungus AT Trees
    DESCRIPTION
          "Some kind of vaguely familiar fungus is growing here on a vine."
    VERB Take DOES
         LOCATE OBJECT IN inventory.
          "Taken!"
    END VERB.
    VERB Drop DOES
         LOCATE OBJECT HERE.
         "The fungus immediately clings to a new vine."
    END VERB.
    VERB eat DOES
         "You try a bit of the fungus. It doesn't taste bad although it
isn't that delicious either. You swallow the rest of it almost
         without chewing. After a short while a strange odour starts
         perspiring from your body."
LOCATE fungus AT nowhere.
         MAKE hero repelling.
    END VERB.
END OBJECT.
_____
SYNTAX take_inventory = 'inventory'.
SYNONYMS i = 'inventory'.
VERB take_inventory DOES
    LIST inventory.
END VERB.
SYNTAX 'look' = 'look'.
SYNONYMS 1 = 'look'.
VERB 'look' DOES
    LOOK.
END VERB.
SYNTAX 'quit' = 'quit'.
SYNONYMS q = 'quit'.
VERB 'quit' DOES
    QUIT.
END VERB.
```

-- NOTE ! It is NOT necessary to declare the actor Hero (which is the -- player himself). But IF you want to make in possible to give -- the Hero certain attributes, THEN you have to declare it. ACTOR Hero IS NOT repelling. END ACTOR. START AT path.

START AT path. "\$p'Oh, my head. It hurts. Why am I out here when I've got this kind of headache? And where is 'here'? And who am I?'"

J COPYING CONDITIONS

The Alan Adventure Development System is REQUEST-WARE. This means that copies of the documentation and executables may be received from the ThoNi&GorFo Adventure Factories, henceforth called The Factories, for free on request.

A copy of these conditions must accompany any copy of the Alan System.

J.1 Documentation

The documentation is copyrighted by The Factories. Copying is allowed provided it is distributed as a whole, or quoted accompanied with appropriate references. This includes putting the documentation (in Postscript or plain text) on FTP-sites, BBS'es or other public access computer networks or nodes, in fact this is encouraged as it will promote the usage of the Alan System. The source text for any example adventures are subject to the same conditions.

J.2 Executables

The Alan system contains two executable programs, the compiler Alan and the interpreter Arun.

Distribution of the interpreter alone or together with game data produced by the compiler is allowed without restrictions or royalty claims provided appropriate references and acknowledgment accompanies the game in documentation or program output. In addition a description of the game, its plot and major features, and/or the game itself (preferably in source) should be donated to The Factories. The Factories agree to any copying or copyright restrictions placed on such a game.

The compiler may not be distributed without registering the receiver with The Factories. Furthermore placing the compiler on FTP-sites, BBS'es or other public access computer networks or nodes is prohibited. Instead persons registering a request with The Factories will receive a free copy of the latest version and the right to use the compiler and its output as described above.

Registered users will receive free notification of updates, new platforms supported, information on commercially or otherwise released games and other information supplied by other users or The Factories.

Registration is free and preferably made through a simple email message. Requesting a distribution through email will automatically register the requestor.

J.3 Source

No source will ever be placed in the public domain or otherwise disclosed to third parties except for porting to new platforms.

J.4 Examples

The Factories would appreciate any example adventures or solutions to problems to improve the documentation and user support. However Alan source marked as such will be considered not copyrighted and may or may not be used, as a whole or in part, in the Alan documentation or distributed in other forms by decision of The Factories.

J.5 Versions, compatibility and support

The Alan System is versioned using a three level number coding scheme, indicating version number, release and correction respectively. Major differences in the language or the introduction of many new features will be indicated by an increment of the version number. Minor changes to the language and introduction of features are indicated by an increment of the release number. Bug fixes will increase the correction number.

Any adventure files and interpreters having the same version and release numbers will be compatible. Adventure files are also compatible across all supported platforms, this includes character sets so any multinational characters are correctly presented on any system. This gives complete coverage of the supported platforms from a single machine.

As the Alan System is a non-profit project user support may vary. To maximise probability of handling, error reports should be sent to The Factories (preferably by email) and include source, version of compiler and interpreter as well as a detailed description of how to reproduce the error and its symptoms.

Corrections and releases will be issued on irregular intervals, restricted to registered users.

J.6 Distribution

The Alan System is mainly distributed through electronic mail. This distribution is free.

Physical media, such as disk or tape, may be supported depending on platform. A requirement is that the requester supplies appropriate media. The cost for physical media distribution may vary.

J.7 Executive Summary

So, in short, the interpreter Arun and any game produced using the Alan System is yours. You may sell or copy it as you like, and as you need the interpreter to run the game it may be copied freely too. The Arun interpreter may also be uploaded on BBS'es or FTP-sites to allow players to download an interpreter for his platform and use that to run your game.

The documentation and examples are free to copy or place on any BBS'es or FTP-sites if their contents are not changed.

The Alan compiler may only be possessed by registered users. To register means you tell The Factories who you are, preferably an email address, but postal mail is fine too. No payment is required. So rather that copying the System for your friends tell them how to contact us, and we'll send them a copy.

Distribution on disks or tape may cost depending on the media. A floppy disc distribution is free, provided you supply the disk.

If you find the Alan Compiler on a BBS or an FTP-site, we would like your help to notify us and the responsible persons that they are breaking the copying conditions of the Alan System. Instead a note or the Alan documentation could be uploaded.

If you create a game using the Alan System we like to see it. Send us a copy (preferably in source) and any documentation or a description of the game and its novel features. This will also add to the suite of test data and therefore improve the quality of future releases as well as allow us to find and document any incompatibilities. If you also enclose a solution we can automate the testing even further. The Factories will not redistribute your game without your written permission.

Short games or samples of Alan source are welcome as examples that we might use and distribute to other users. Sending an example means you waive all rights to it.

Α

Abug 72 ACTOR 16, 39 in what specifications 49 actor 12 actors execution context 61 hints about 68 moving 61 adjective 34 AFTER qualifier 30 aggregate COUNT 52 MAX 52 SUM 52 ALL 26, 36, 60, 79, 80 alternatives, verb 29 AND 26, 59, 60 article 35 Arun 71, 79, 102, 103 AT 49 attributes 15, 52, 63 boolean 23 default 23 numeric 23 of actors 39 of locations 32 of objects 35 string 23

В

BEFORE qualifier 30 binary operators 51 BNF 85 boolean attributes 23 BUT 60, 79, 80

С

CANCEL statement 46 character combinations, in strings 42 character sets 22 CHECK 28 in exits 32, 61

in verbs 17 check, unconditional 28 checks execution order 31 common verbs 65 comparisons 51 concatenation, in player commands 59 **CONTAINER 37** container property of actors 39 of objects 34 containers closing 66 hints about 66 contexts of execution 60 COUNT aggregate 52 in limits 37

D

Debug option 22 debugging 71 **DECREASE** statement 47 default attributes 23, 65 syntax 19, 27 DESCRIBE statement 36, 43 DESCRIPTION of actor scripts 40, 69 of actors 40 of locations 14, 32 descriptions execution context 60 DOES in exits 61 in locations 32, 61 in verbs 28 doors, hints about 65 double quotes 56

Ε

EMPTY statement 46 equality 51 EVENT 38

Index

events execution context 61 hints about 70 EVERYTHING 60 EXCEPT 60, 79 execution contexts 60 execution of an adventure 59 EXIT 13, 32, 61 expression types 50 expressions 50 logical 51

Η

HEADER 38 HERE 49 hero 41, 61

I

identifiers case translation of 55 lexical definition 55 quoted 55 IF statement 15, 48, 64 IN 49 include construct 57 files 57, 65 INCREASE statement 47 Infocom 9, 10 IT 59, 79

L

Language option 22 languages 79 Length option 22 LIMITS 37 execution of 37, 46 LIST statement 43 literals 52 LOCATE statement 15, 45 locating inside containers 37, 46 LOCATION 31 in what specification 49 location 11, 13 logical expressions 51 logical operators 51 LOOK statement 44

Μ

MAKE statement 15, 47 map 11 MAX aggregate 52 MENTIONED 36 multinational characters 22 multiple indicator 26, 60 multiple parameters 60

Ν

NAME of actors 39 of locations 31, 55 of objects 33 NEARBY 49 noun 34 numbers lexical definition 56 numeric attributes 23

0

OBJECT 14, 33 in what specifications 49 object 11 shadow 67 ONLY qualifier 30 operators binary 51 logical 51 relational 52 options 21 output statements 42

Ρ

Pack option 22 parameter 19, 26 classes 26 referencing 60 player commands 59 presence, of parameters 60 Index

Q

qualifiers, verb 28, 30 QUIT statement 44 quoted identifier 55 quotes double 56 single 56 string 56

R

RANDOM 52 relational operators 52 RESTORE statement 44 restriction, of parameters 19, 26 rules 41 executing 41 execution context 61

S

SAVE 80 SAVE statement 44 SAY statement 43 SCHEDULE statement 46 SCORE statement 44 SCRIPT 40 scripts description of 40 SET statement 15, 47 shadow object 67 single quotes 56 spacing, in strings 56 start section 21, 41 statements, output from 42 STEP 40 step, executing the last 41 STRING 14, 42 string attributes 23 comparisons 51 special character combinations 42 string quotes 56 strings lexical definition 56

spacing 56 SUM aggregate 52 SYNONYMS 24 SYNTAX 19, 26 syntax, default 19, 27

Т

THEM 60, 79 THEN 59 types of expressions 50

U

unconditional check 28 USE statement 40, 48

V

VERB 16, 28 global 17 in location 17 in object 17 verb 12 alternative 29 execution context 60 execution order 18, 20, 30 qualifiers 28, 30 reusing common 65 VISITS statements 45

W

what specifications 49 WHEN 41 where specification 48 Width option 22 126